


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L. 1.000

PER IL TUO PERSONAL COMPUTER

Una pubblicazione della J.soft editrice



PAPPER

soft

4

Anno 2 - N° 4 - 1 febbraio 1985



Robot



**Videogames I
Escher**



**Ring
Scappa...
Demo L/M**



**Database
Dots**



**Crash
Disegnatore
Fotoreporter**

Guida all'input dei programmi sullo ZX Spectrum

Le "parole" comprese tra parentesi graffe indicano i caratteri grafici predefiniti (G), il tasto (numero seguente la G), la necessità di premere il tasto insieme a CAPS SHIFT (eventuale S precedente la G) e il numero di ripetizioni del tasto richieste (eventuale numero all'inizio della "parola"). I caratteri grafici definiti dal programma sono invece indicati da lettere maiuscole (corrispondenti al tasto da usare) sottolineate.

Quando leggete	Premete	Vedrete
{G1}	1	█
{G2}	2	█
{G3}	3	█
{G4}	4	█
{G5}	5	█
{G6}	6	█
{G7}	7	█
{G8}	8	█
{SG1}	CAPS SHIFT 1	█
{SG2}	CAPS SHIFT 2	█
{SG3}	CAPS SHIFT 3	█
{SG4}	CAPS SHIFT 4	█
{SG5}	CAPS SHIFT 5	█
{SG6}	CAPS SHIFT 6	█
{SG7}	CAPS SHIFT 7	█
{SG8}	CAPS SHIFT 8	█

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Quando leggete	Premete	Vedrete
A	A	Simbolo grafico definito nel programma in uso.
B	B	
C	C	
D	D	
E	E	
F	F	
G	G	
H	H	
I	I	
J	J	
K	K	
L	L	
M	M	
N	N	
O	O	
P	P	
Q	Q	
R	R	
S	S	
T	T	
U	U	

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono "parole" racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da [< >] deve essere




premutato contemporaneamente al tasto COMMODORE. Inoltre, se tra parentesi si trova un carattere alfabetico "solitario", questo dovrà essere premuto contemporaneamente al tasto CONTROL.

Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

{CLR}	SHIFT CLR/HOME	█	{CYN}	CTRL 4	█	[< 7 >]	G 7	█
{HOME}	CLR/HOME	█	{PUR}	CTRL 5	█	[< 8 >]	G 8	█
{SU}	SHIFT ↑ CRSR ↓	█	{GRN}	CTRL 6	█	{F1}	f1	█
{GIU'}	↑ CRSR ↓	█	{BLU}	CTRL 7	█	{F2}	f2	█
{SIN}	SHIFT ← CRSR →	█	{YEL}	CTRL 8	█	{F3}	f3	█
{DES}	← CRSR →	█	[< 1 >]	G 1	█	{F4}	f4	█
{RVS}	CTRL 9	█	[< 2 >]	G 2	█	{F5}	f5	█
{OFF}	CTRL 0	█	[< 3 >]	G 3	█	{F6}	f6	█
{BLK}	CTRL 1	█	[< 4 >]	G 4	█	{F7}	f7	█
{WHT}	CTRL 2	█	[< 5 >]	G 5	█	{F8}	f8	█
{RED}	CTRL 3	█	[< 6 >]	G 6	█			█



PAPER SOFT

	 Apple II	4	Robot di B. Smith trad. e adatt. di M. Cerofolini
	 TI-99/4A	7	Videogames I di H. Renko e S. Edwards trad. e adatt. di E. Re Garbagnati
	 TI-99/4A	10	Escher di H. Renko trad. e adatt. di E. Re Garbagnati
16/48K	 Sinclair Spectrum	12	Ring di M.J. Bradford trad. e adatt. di C. Panzalis
16/48K	 Sinclair Spectrum	15	Scappa... di D. Creighton trad. e adatt. di C. Panzalis
	C64	18	Database di R. Christensen trad. e adatt. di F. Sarcina
	C64	23	Dots di J. Arnold trad. e adatt. di F. Sarcina
	VIC-20	24	Crash di D. Liu trad. e adatt. di F. Sarcina
Joystick opzionale	VIC-20	27	Disegnatore di J. Wise trad. e adatt. F. Sarcina
Joystick	VIC-20	28	Fotoreporter di C.D. Lane trad. e adatt. di F. Sarcina
16/48K	 Sinclair Spectrum	30	Demo L/M di G. Locke trad. e adatt. di C. Panzalis

J. soft s.r.l.

**DIREZIONE, REDAZIONE,
AMMINISTRAZIONE**

Via Rosellini, 12
20124 MILANO
Tel. (02) 68.88.228-68.37.97

DIRETTORE RESPONSABILE:

Pietro Dell'Orco

COORDINAMENTO TECNICO:

Riccardo Paolillo

REDAZIONE:

Lucio Bragagnolo
Mauro Cristuib. Grizzi

GRAFICA E IMPAGINAZIONE:

Margherita La Noce
Ivana Rossi
Raffaella Toffolatti

FOTOCOMPOSIZIONE:

d&b Via Vignola, 5
Tel. 02/59.85.08
20133 MILANO

CONTABILITÀ:

Giulia Pedrazzini
Flavia Bonaiti

**AUTORIZZAZIONE ALLA
PUBBLICAZIONE:**

Tribunale di Milano n° 200
del 14.04.1984

STAMPA:

Elcograf - Beverate (CO)



Rivista associata
all'Unione stampa
Periodica Italiana

PUBBLICITÀ

Concessionario per l'Italia e l'Estero
J Advertising s.r.l.,
V.le Restelli, 5
20124 MILANO
Tel. (02) 68.82.895-68.80.606-68.87.233
Tlx. 316213 REINA I
Concessionario esclusivo per la
DIFFUSIONE in Italia e Estero:
SODIP - Via Zuretti, 25
20125 MILANO
Spedizione in abbonamento
postale Gruppo II/70
Prezzo della rivista L. 1.000
Numeri arretrati L. 2.000

© TUTTI I DIRITTI DI
RIPRODUZIONE O TRADUZIONE
DEGLI ARTICOLI PUBBLICATI
SONO RISERVATI



Robot

Con questo programma vi verrà presentato un personaggio, il Robot, che potrete animare a vostro piacimento.

Le istruzioni che potete dargli sono semplici, ma dovrete esercitarvi un po', prima di acquisire l'abilità necessaria per farlo marciare nella direzione desiderata, ed in seguito per riuscire a terminare con successo il gioco proposto. Questo consiste nel guidare il robot in modo che calpesti con il calcagno tutti e tre i quadrati bianchi disposti sullo schermo; al termine il computer vi renderà noto il tempo impiegato, così potrete sfidare un amico in una divertente gara di precisione e di velocità. I tasti

che vi permetteranno di trasmettere gli ordini al Robot sono:

F voltati di fronte
L voltati a sinistra
R voltati a destra
M marcia in avanti
C sali
D scendi
H fermati
Y annuisci
N accenna di no col capo
B batti le palpebre
P inizia la fase di gioco.

```
10 GOTO 9000
20 KEY = PEEK ( - 16384): IF KEY
  < 128 THEN 60
25 PRINT : PRINT : PRINT
26 IF FP = 1 THEN FP = 2
30 IF KEY < > 194 AND KEY < >
  198 AND KEY < > 204 AND KEY
  < > 206 AND KEY < > 210 AND
  KEY < > 217 AND KEY < > 19
  5 AND KEY < > 196 AND KEY <
  > 205 AND KEY < > 200 AND
  KEY < > 209 AND KEY < > 20
  8 THEN 50
40 A = KEY
50 POKE - 16368,0
60 GOSUB 3000
65 IF FP = 2 THEN GOSUB 4000
70 POKE 60,0: POKE 61,4: POKE 62
  ,255: POKE 63,7: POKE 66,0: POKE
  67,8:
72 REM CALL-468:POKE -16299,0
80 GR
90 IF X < 6 OR X > 33 THEN 93
91 IF Y < 5 OR Y > 24 THEN 96
92 GOTO 20
93 GOSUB 8010: GOSUB 8000: IF X <
  6 THEN X = X + 1
94 IF X > 33 THEN X = X - 1
96 IF Y < 5 THEN Y = Y + 1
97 IF Y > 24 THEN Y = Y - 1
98 GOTO 20
100 COLOR= 4: REM CORPO
110 VLIN Y + 1,Y + 7 AT X: VLIN
  Y + 1,Y + 7 AT X - 1: VLIN Y
  + 1,Y + 7 AT X + 1
```

```
120 RETURN
200 COLOR= 15: REM TESTA FRONT
  ALE
210 PLOT X,Y: HLIN X - 1,X + 1 AT
  Y - 1: PLOT X - 2,Y - 2: PLOT
  X - 1,Y - 2: PLOT X + 1,Y -
  2: PLOT X + 2,Y - 2: PLOT X -
  2,Y - 3: PLOT X,Y - 3: PLOT
  X + 2,Y - 3
220 HLIN X - 1,X + 1 AT Y - 4
230 RETURN
250 COLOR= 15: REM CHIUDE GLI O
  CCHI
260 PLOT X,Y: HLIN X - 1,X + 1 AT
  Y - 1: HLIN X - 2,X - 1 AT Y
  - 2: HLIN X - 2,X + 2 AT Y -
  3: HLIN X - 1,X + 1 AT Y - 4
  : HLIN X + 1,X + 2 AT Y - 2
270 RETURN
300 COLOR= 15: REM VISO DI FRO
  NTE CHE GUARDA IN BASSO
310 HLIN X - 1,X + 1 AT Y: PLOT
  X - 1,Y - 1: PLOT X + 1,Y -
  1: PLOT X - 2,Y - 2: PLOT X,
  Y - 2: PLOT X + 2,Y - 2: HLIN
  X - 1,X + 1 AT Y - 3: HLIN X
  - 1,X + 1 AT Y - 4
320 RETURN
400 COLOR= 15: REM TESTA A SIN
  ISTRÀ
410 PLOT X,Y: HLIN X - 1,X + 1 AT
  Y - 1: HLIN X - 2,X + 1 AT Y
  - 2: PLOT X + 1,Y - 3: HLIN
  X,X + 1 AT Y - 4: PLOT X - 1
  ,Y - 3
```



```

420 RETURN
500 COLOR= 15: REM TESTA A DEST
RA
510 PLOT X,Y: HLIN X - 1,X + 1 AT
Y - 1: HLIN X - 1,X + 2 AT Y
- 2: PLOT X - 1,Y - 3: HLIN
X - 1,X AT Y - 4: PLOT X + 1
,Y - 3
520 RETURN
900 COLOR= 11: REM BRACCIA E GA
MBE PER LA FACCIA A DESTRA
910 VLIN Y + 8,Y + 14 AT X: VLIN
Y + 1,Y + 7 AT X: PLOT X + 1
,Y + 14
920 RETURN
950 COLOR= 11: REM BRACCIA E GA
MBE PER FACCIA A SINISTRA
960 VLIN Y + 8,Y + 14 AT X: VLIN
Y + 1,Y + 7 AT X: PLOT X - 1
,Y + 14
970 RETURN
1000 REM COLOR=11:REM BRACCIA E
GAMBE FRONTALI
1005 COLOR= 11
1010 PLOT X - 2,Y + 1: PLOT X +
2,Y + 1:
1015 VLIN Y + 2,Y + 7 AT X + 3: VLIN
Y + 2,Y + 7 AT X - 3
1020 VLIN Y + 8,Y + 14 AT X - 1:
VLIN Y + 8,Y + 14 AT X + 1
1050 RETURN
1100 REM BRACCIA E GAMBE CHE MA
RCIANO A SINISTRA
1110 COLOR= 11
1120 PLOT X,Y + 1: PLOT X - 1,Y +
2: PLOT X - 2,Y + 3: PLOT X -
3,Y + 4: PLOT X - 4,Y + 5: PLOT
X + 2,Y + 3
1130 VLIN Y + 8,Y + 14 AT X: VLIN
Y + 10,Y + 13 AT X - 2: PLOT
X - 1,Y + 9: PLOT X - 3,Y +
13: PLOT X - 1,Y + 14
1140 RETURN
1150 REM BRACCIA E GAMBE CHE MA
RCIANO A SINISTRA 1
1160 COLOR= 11
1170 PLOT X,Y + 1: PLOT X + 1,Y +
2: PLOT X + 2,Y + 3: PLOT X +
1,Y + 4: PLOT X,Y + 5: PLOT
X - 2,Y + 3: PLOT X - 3,Y +
4: PLOT X - 4,Y + 5
1180 VLIN Y + 8,Y + 14 AT X: PLOT
X + 1,Y + 11: PLOT X - 1,Y +
14: PLOT X + 1,Y + 13: PLOT
X + 2,Y + 12: PLOT X - 1,Y +
10
1190 RETURN
1200 REM BRACCIA E GAMBE CHE MA
RCIANO A DESTRA
1210 COLOR= 11
1220 PLOT X,Y + 1: PLOT X + 1,Y +
2: PLOT X + 2,Y + 3: PLOT X +
3,Y + 4: PLOT X + 4,Y + 5: PLOT
X - 2,Y + 3
1230 VLIN Y + 8,Y + 14 AT X: VLIN
Y + 10,Y + 13 AT X + 2: PLOT
X + 1,Y + 9: PLOT X + 3,Y +
13: PLOT X + 1,Y + 14
1240 RETURN
1250 REM BRACCIA E GAMBE CHE MA
RCIANO A SINISTRA 1
1260 COLOR= 11
1270 PLOT X,Y + 1: PLOT X - 1,Y +
2: PLOT X - 2,Y + 3: PLOT X -
1,Y + 4: PLOT X,Y + 5: PLOT
X + 2,Y + 3: PLOT X + 3,Y +
4: PLOT X + 4,Y + 5
1280 VLIN Y + 8,Y + 14 AT X: PLOT
X - 1,Y + 11: PLOT X + 1,Y +
14: PLOT X - 1,Y + 13: PLOT
X - 2,Y + 12: PLOT X + 1,Y +
10
1290 RETURN
1940 D = 0: REM BLINK
1941 IF F > 1 THEN F = 0
1942 GOSUB 100: GOSUB 1000
1943 IF F = 0 THEN 1946
1945 GOSUB 250:F = 0: RETURN
1946 GOSUB 200:F = 1: RETURN
1950 REM SALIRE
1951 Y = Y - 1: GOTO 2051
1960 REM SCENDERE
1961 Y = Y + 1: GOTO 2051
1980 D = 0: REM FRONTE
1981 GOSUB 200: GOSUB 100: GOSUB
1000: RETURN
2000 REM HALT
2001 IF D = 1 THEN A = 210
2002 IF D = 2 THEN A = 204
2003 POP : POP : GOTO 20
2040 D = 2: REM FACCIA A DESTRA
2041 GOSUB 400: GOSUB 100: GOSUB
950: RETURN
2050 REM MARCIA AVANTI
2051 GOSUB 500
2052 RETURN
2060 D = 0: REM MUOVE LA TESTA
2061 GOSUB 100: GOSUB 1000: IF F
= 0 THEN 2064
2062 IF F = 2 THEN 2065
2063 F = F1: GOSUB 200: RETURN
2064 F = 1:F1 = 2: GOSUB 400: RETURN
2065 F = 1:F1 = 0: GOSUB 500: RETURN
2080 GOSUB 100: GOSUB 200: GOSUB
1000: COLOR= 15: PLOT 10,26:
PLOT 20,38: PLOT 30,24
2081 PRINT : PRINT "FAI MARCIARE
ROM E FAI ANDARE IL SUO
TALLONE SOPRA AI TRE QUADRAT
I BINACHI A TURNO"
2083 C = 0:S1 = 0:S2 = 0:S3 = 0:F
P = 1: RETURN
2090 REM ROUTINE DI FINE
2092 POKE - 16300,0: TEXT : HOME
2094 VTAB 10: HTAB 18: PRINT "CI
AO!"
2096 POKE 75,8: POKE 205,8: END
2100 D = 1: REM FACCIA A SINISTR
A
2101 GOSUB 500: GOSUB 100: GOSUB
900: RETURN
2170 D = 0: REM ANNUISCE

```



```

2171 IF F > 1 THEN F = 0
2172 GOSUB 100: GOSUB 1000: IF F
= 0 THEN 2174
2173 GOSUB 200: F = 0: RETURN
2174 GOSUB 300: F = 1: RETURN
3000 REM GOTO CALCOLATO
3010 IF A = 194 THEN GOSUB 1940
: RETURN
3020 IF A = 198 THEN GOSUB 1980
: RETURN
3030 IF A = 204 THEN GOSUB 2040
: RETURN
3040 IF A = 206 THEN GOSUB 2060
: RETURN
3050 IF A = 210 THEN GOSUB 2100
: RETURN
3060 IF A = 217 THEN GOSUB 2170
: RETURN
3070 IF A = 195 THEN GOSUB 1950
: RETURN
3080 IF A = 196 THEN GOSUB 1960
: RETURN
3090 IF A = 205 THEN GOSUB 2050
: RETURN
3100 IF A = 200 THEN GOSUB 2000
: RETURN
3110 IF A = 209 THEN GOSUB 2090
: RETURN
3120 IF A = 208 THEN GOSUB 2080
: RETURN
3130 STOP
4000 REM GIOCO DELLA PARATA
4010 IF S1 AND S2 AND S3 THEN 41
40
4020 COLOR= 15: C = C + 1: IF C =
999 THEN 4140
4028 VTAB 21
4029 PRINT : PRINT "F=FRONTE R=D
ESTRA L=SINISTRA M=AVANTI
C=SALE D=SCENDE H=ALT"
4030 VTAB 23: HTAB 29: PRINT "PU
NTI "; C
4040 IF S1 = 1 THEN 4060
4050 PLOT 10,26: IF X = 10 AND Y
= 11 THEN GOSUB 4110
4060 IF S2 = 1 THEN 4080
4070 PLOT 20,38: IF X = 20 AND Y
= 23 THEN GOSUB 4120
4080 IF S3 = 1 THEN 4100
4090 PLOT 30,24: IF X = 30 AND Y
= 9 THEN GOSUB 4130
4100 RETURN
4110 S1 = 1: RETURN
4120 S2 = 1: RETURN
4130 S3 = 1: RETURN
4140 : PRINT : PRINT : PRINT "IL
TUO PUNTEGGIO DI "; C: IF C >
160 THEN 4240
4141 IF C > 120 THEN 4220
4142 IF C > 80 THEN 4200
4143 IF C > 40 THEN 4180
4160 PRINT "E' ECCELLENTE.": PRINT
"ARRUOLATI!!": GOTO 4250

```

```

4180 PRINT "E' BUONO "; B$: PRINT
"MA SI PUO' MIGLIORARE": GOTO
4250
4200 PRINT "E' MEDIO "; B$: PRINT
"CONTINUA AD ESERCITARTI": GOTO
4250
4220 PRINT "E' BASSO "; B$: PRINT
"PROVACI ANCORA": GOTO 4250
4240 PRINT "E' PESSIMO "; B$: PRINT
"DEVI PROPRIO MIGLIORARE": GOTO
4250
4245 FP = 0
4250 PRINT "PER GIOCARE ANCORA P
REMI 'P', PER FINIRE 'Q'";
4260 RETURN
5000 REM MARCIA AVANTI
5010 IF D = 1 THEN 5040
5020 IF D = 2 THEN 5090
5030 POP : POP : POP : A = 206: GOTO
20
5040 X = X + 1: GOSUB 500: GOSUB
100: IF F = 0 THEN 5070
5050 IF F = 2 THEN 5080
5060 F = F1: GOSUB 900: RETURN
5070 F = 1: F1 = 2: GOSUB 1200: RETURN
5080 F = 1: F1 = 0: GOSUB 1250: RETURN
5090 X = X - 1: GOSUB 400: GOSUB
100: IF F = 0 THEN 5120
5100 IF F = 2 THEN 5130
5110 F = F1: GOSUB 950: RETURN
5120 F = 1: F1 = 2: GOSUB 1100: RETURN
5130 F = 1: F1 = 0: GOSUB 1150: RETURN
5140 RETURN
8000 REM ROUTINE GRUNT
8002 FOR Z = 1 TO 10: ZZ = PEEK
(- 16336): FOR Z2 = 1 TO 2:
NEXT Z2: NEXT Z: RETURN
8010 PRINT "AH!!! HO SBATTUTO I
L MIO NASO!!"
8011 RETURN
9000 REM INIZIALIZZAZIONE
9010 POKE 74,0: POKE 75,12
9020 POKE 204,0: POKE 215,12
9030 X = 19: Y = 19
9050 FOR A = 20 TO 24: VTAB A: PRINT
": NEXT A
9068 A = 198: GR
9070 GOTO 20
9999 END

```

**Aspettiamo
i tuoi lavori
migliori!**

Videogames I

La lentezza del BASIC del nostro TI-99 avrà già fatto esclamare a più d'uno tra voi: "Ah, se potessi poter caricare una volta per tutte nel computer i miei programmi di giochi, senza dover attendere interminabili minuti per la lettura dei files da registratore!..."

Ebbene, battendo questo semplice listato (naturalmente in BASIC), avrete a disposizione non il solito giochino, ma ben tre videogames delle più svariate categorie. Dopo aver dato il RUN, infatti, vi verrà offerto il menu principale (ed unico...) dal quale potrete scegliere di buttarvi nella fantascienza con Galactic Monsters. Dovrete destreggiarvi, in questo gioco, a superare incolume la fascia di Van Allen (che non ha nulla a che vedere con il quasi omonimo gruppo rock (ma non è altro che il nome di una delle innumerevoli zone enigmatiche dell'universo), popolata da mostri dai quali l'unica possibilità di difesa è evitarli. Attenzione, però a non farvi beccare più di 5 volte...

Tornati al menu potrete quindi scegliere un gioco cerebrale sul tipo del cubo di Rubick; One to five è infatti un gioco "exasperante" che,

a quanto mi risulta, è stato inventato dalla CASIO per un suo calcolatore. Vi verrà mostrato un quadrato composto da 9 cifre casuali comprese tra 0 e 5. Scopo del gioco è fare in modo che tutte le 9 cifre diventino 0. Per questo dovrete premere i tasti visualizzati sulla destra dello schermo: ognuno di essi agirà su 5 cifre aumentandole di 1, in modo che premendolo più volte si avrà un ciclo numerico di questo tipo: 0-1-2-3-4-5-0-1-... Comunque fate un paio di partite ed imparerete subito.

L'ultimo gioco compreso è Shark Hunt: vi troverete a bordo del vostro Yacht, a caccia di un famigerato squalo, equipaggiati del vostro solo sonar e del vostro inseparabile arpione. Il sonar vi guiderà verso lo squalo emettendo suoni sempre più acuti a seconda della vicinanza alla preda. Potrete effettuare i vostri movimenti con i soliti tasti con le frecce, e sparare premendo la barra.

In ogni gioco è previsto che, premendo il tasto -T-, in qualsiasi momento si ritorni al Menu. Inutile aggiungere altro: Buona Fortuna!!!

```

100 REM *****
110 REM ** VIDEOGAMES-1- **
120 REM *****
130 REM TI-99/4A          BASIC
140 CALL CLEAR
150 CALL SCREEN(10)
160 PRINT "#####
#####"
170 PRINT TAB(7);"VIDEOGAMES -1-
"
180 PRINT "#####
#####"
190 PRINT :::::~::~:
-----"
200 PRINT :::"MENU:  1. GALACTIC
MONSTERS          2. ONE TO FIVE

```

```

3. SHARK HUNT
4. FINE GIOCHI"::
210 PRINT "-----
-----"::
220 CALL KEY(0,E,R)
230 IF R=0 THEN 220
240 IF E=50 THEN 1580
250 IF E=51 THEN 2030
260 IF E=52 THEN 2970
270 IF E<>49 THEN 220
280 REM GALACTIC MONSTERS
290 CALL CLEAR
300 A1$="0066421818423C00"
310 A2$="007E7E7E7E7E7E00"
320 A3$="18183C5A993C2466"
330 CALL COLOR(1,8,16)
340 CALL COLOR(2,9,16)
350 CALL COLOR(3,2,16)
360 CALL CHAR(35,A1$)
370 CALL CHAR(42,A2$)

```



```

380 CALL CHAR(55,A3$)
390 DIM X(12)
400 DIM Y(12)
410 GOTO 1360
420 REM SCHERMO*****
430 CALL CLEAR
440 FOR T=1 TO 11
450 CALL HCHAR(5+T,10,45,11)
460 NEXT T
470 CALL SOUND(300,262,0)
480 CALL SOUND(300,330,0)
490 CALL SOUND(300,392,0)
500 CALL SOUND(1000,262,0,330,0,
392,0)
510 CALL HCHAR(13,10,35,11)
520 CALL HCHAR(6,15,55)
530 X1=6
540 Y1=15
550 FOR T=1 TO 11
560 X(T)=13
570 Y(T)=9+T
580 NEXT T
590 RETURN
600 REM MOSSA GIOCATORE***
610 CALL SOUND(10,880,0)
620 CALL KEY(3,KEY,ST)
630 CALL HCHAR(X1,Y1,45)
640 TURNS=URNS+1
650 IF KEY=88 THEN 710
660 IF KEY=83 THEN 740
670 IF KEY=68 THEN 770
680 IF KEY=69 THEN 800
690 IF KEY=84 THEN 140
700 GOTO 820
710 IF X1=16 THEN 820
720 X1=X1+1
730 GOTO 820
740 IF Y1=10 THEN 820
750 Y1=Y1-1
760 GOTO 820
770 IF Y1=20 THEN 820
780 Y1=Y1+1
790 GOTO 820
800 IF X1=6 THEN 820
810 X1=X1-1
820 CALL HCHAR(X1,Y1,55)
830 RETURN
840 REM CATTURA*****
850 FOR T=1 TO 11
860 IF X1<>X(T) THEN 930
870 IF Y1<>Y(T) THEN 930
880 CALL SOUND(-100,440,0)
890 CALL SOUND(-100,220,0)
900 CALL SOUND(-100,110,0)
910 CALL SOUND(1000,-4,0)
920 K=K+1
930 NEXT T
940 RETURN
950 REM MOSSA MOSTRI*****
960 FOR T=1 TO 11
970 CALL HCHAR(X(T),Y(T),45)
980 IF RND<CHANCE THEN 1240

```

```

990 A=INT(RND*4)
1000 IF A=0 THEN 1040
1010 IF A=1 THEN 1060
1020 IF A=2 THEN 1080
1030 GOTO 1100
1040 X(T)=X(T)+1
1050 GOTO 1110
1060 X(T)=X(T)-1
1070 GOTO 1110
1080 Y(T)=Y(T)+1
1090 GOTO 1110
1100 Y(T)=Y(T)-1
1110 IF X(T)<6 THEN 1160
1120 IF X(T)>16 THEN 1180
1130 IF Y(T)<10 THEN 1200
1140 IF Y(T)>20 THEN 1220
1150 GOTO 1230
1160 X(T)=X(T)+1
1170 GOTO 1300
1180 X(T)=X(T)-1
1190 GOTO 1300
1200 Y(T)=Y(T)+1
1210 GOTO 1300
1220 Y(T)=Y(T)-1
1230 GOTO 1300
1240 X2=X1-X(T)
1250 Y2=Y1-Y(T)
1260 IF RND<.8 THEN 1290
1270 X(T)=X(T)+SGN(X2)
1280 GOTO 1300
1290 Y(T)=Y(T)+SGN(Y2)
1300 CALL HCHAR(X(T),Y(T),35)
1310 NEXT T
1320 FOR U=1 TO 11
1330 CALL HCHAR(X(U),Y(U),35)
1340 NEXT U
1350 RETURN
1360 REM PROGRAMMA *****
      PRINCIPALE*****
1370 CHANCE=CHANCE+.1
1380 GOSUB 420
1390 GOSUB 600
1400 GOSUB 840
1410 GOSUB 950
1420 IF K>4 THEN 1510
1430 IF X1<16 THEN 1390
1440 SCORE=SCORE+POINTS-TURNS-5*
K
1450 POINTS=POINTS+20
1460 TURNS=0
1470 GOTO 1370
1480 CALL SOUND(-100,110,0)
1490 CALL SOUND(1000,-4,0)
1500 RETURN
1510 PRINT "Questo gioco e'finit
o."
1520 PRINT "Il tuo punteggio e'
"
1530 PRINT SCORE
1540 PRINT "Arrivederci_(premi u
n tasto)"
1550 CALL KEY(0,K,S)

```



```

1560 IF S=0 THEN 1550
1570 GOTO 140
1580 REM. ONE TO FIVE*****
1590 CALL CLEAR
1600 RANDOMIZE
1610 IL$="234WERASD"
1620 FOR F3=0 TO 2
1630 FOR F4=0 TO 2
1640 CALL HCHAR(F3+10,F4+13,ASC(
STR$(H(F3,F4))))
1650 CALL HCHAR(F3+10,F4+19,ASC(
SEG$(IL$,3*F3+F4+1,1)))
1660 NEXT F4
1670 NEXT F3
1680 FOR F1=0 TO 10
1690 I=INT(RND*9)+1
1700 GOSUB 1800
1710 NEXT F1
1720 M=0
1730 CALL KEY(1,I,B)
1740 IF (I<1)+(I>9)*(I<>11)<>0 T
HEN 1730
1750 IF B=0 THEN 1730
1760 IF I=11 THEN 140
1770 M=M+1
1780 GOSUB 1800
1790 GOTO 1730
1800 W=INT((9-I)/3+1)
1810 Z=((I-1)/3-INT((I-1)/3))*3+
1
1820 FOR F2=0 TO 2
1830 H(W-1,F2)=H(W-1,F2)+1
1840 H(F2,Z-1)=H(F2,Z-1)+1
1850 NEXT F2
1860 H(W-1,Z-1)=H(W-1,Z-1)-1
1870 F=0
1880 FOR F3=0 TO 2
1890 FOR F4=0 TO 2
1900 IF H(F3,F4)<>6 THEN 1920
1910 H(F3,F4)=0
1920 IF H(F3,F4)<>0 THEN 1940
1930 F=F+1
1940 CALL HCHAR(F3+10,F4+13,ASC(
STR$(H(F3,F4))))
1950 NEXT F4
1960 NEXT F3
1970 IF F<9 THEN 1980 ELSE 2000

1980 RETURN
1990 PRINT
2000 PRINT "SEI GRANDE!!"
2010 PRINT "RISOLTO IN ";M;"MOSS
E"
2020 GOTO 140
2030 REM SHARK HUNT
2040 CALL CLEAR
2050 DIM CO$(3)
2060 NG=0
2070 TR=0
2080 MR=0
2090 FOR COM=1 TO 3
2100 READ CO$(COM)

```

```

2110 NEXT COM
2120 DATA SPIACENTE: TROPPO LONT
ANO
2130 DATA CE L'HAI FATTA!!!
2140 DATA HAI ESAURITO LE ENERGI
E
2150 GOTO 2860
2160 REM INIZIALIZZAZIONE**
2170 CALL CLEAR
2180 PRINT " WER"
2190 PRINT " S D"
2200 PRINT " ZXC"
2210 FOR Z=1 TO 8
2220 PRINT
2230 NEXT Z
2240 NG=NG+1
2250 SX=INT(RND*70-35)
2260 SY=INT(RND*70-35)
2270 DI=ABS(SX)+ABS(SY)
2280 SD=DI
2290 MD=DI
2300 DATA 200,0,0,0,0
2310 READ EN,HI,NS,VX,VY
2320 RETURN
2330 REM SCELTA VELOCITA***
2340 CALL KEY(3,P,U)
2350 VX=3*((P=87)+(P=83)+(P=90)-
(P=82)-(P=68)-(P=67))
2360 VY=3*((P=90)+(P=88)+(P=67)-
(P=87)-(P=69)-(P=82))
2370 SH=(P=32)
2380 IF P=84 THEN 140
2390 RETURN
2400 REM CALCOLO*****

2410 SX=SX-VX
2420 SY=SY-VY
2430 SX=SX+SGN(SX)
2440 SY=SY+SGN(SY)
2450 EN=EN+(VX<>0)+(VY<>0)-1
2460 IF SH=0 THEN 2500
2470 NS=NS+1
2480 HI=(RND*DI<2)
2490 EN=EN-2*DI
2500 EN=- (EN>0)*EN
2510 IF HI THEN 2540
2520 DI=ABS(SX)+ABS(SY)
2530 MD=- (MD<DI)*MD- (MD>=DI)*DI
2540 RETURN
2550 IF SH=0 THEN 2580
2560 CALL SOUND(100,110,0)
2570 GOTO 2590
2580 CALL SOUND(100,3000-25*DI,0
)
2590 X$=STR$(EN)
2600 D=LEN(X$)
2610 FOR F=1 TO 4
2620 IF F<=D THEN 2630 ELSE 2650

2630 CALL HCHAR(2,14+F,ASC(SEG$(
X$,F,1)))
2640 GOTO 2660

```



```

2650 CALL HCHAR(2,14+F,32)
2660 NEXT F
2670 RETURN
2680 REM FINE GIOCO*****

2690 PRINT C0$(CA)
2700 PRINT "DISTANZA DI PARTENZA
: ";SD;"M"
2710 PRINT "NUMERO SPARI:";NS
2720 PRINT "TI SEI AVVICINATO IN
";MD;" MOSSE."
2730 RE=50-DI
2740 IF HI=0 THEN 2770
2750 PRINT "LO HAI COLPITO DA:";
DI;"M"
2760 RE=50+EN/4+SD/10
2770 TR=TR+RE
2780 IF RE<MR THEN 2800
2790 MR=RE
2800 PRINT "PUNTEGGIO SU 100:"
2810 PRINT "QUESTA PARTITA:";INT
(RE)
2820 PRINT "PERCENTUALE SU";NG;"

```

```

PARTITE:";INT(TR/NG)
2830 PRINT "MAX.:";INT(MR)
2840 RETURN
2850 REM PROGRAMMA*****
      PRINCIPALE*****

2860 GOSUB 2160
2870 GOSUB 2330
2880 GOSUB 2400
2890 GOSUB 2550
2900 CA=-3*(EN<=0)-2*HI-(DI>100)

2910 CA=CA+3*(CA>3)
2920 IF CA=0 THEN 2870
2930 GOSUB 2680
2940 PRINT "PREMI UN TASTO"
2950 CALL KEY(0,KJ,XW)
2960 IF XW=0 THEN 2950 ELSE 140

2970 CALL CLEAR
2980 CALL SCREEN(16)
2990 PRINT "CIAO!!!!!!"
3000 END

```



Escher

Questo programma in TI-BASIC, a prima vista persino banale, avrà modo di darvi particolari soddisfazioni.

Esso si rifà all'opera dell'artista grafico tedesco M.C. Escher (1902-1972).

Un'opera basata sulla simmetria e sulle forme matematiche che si appella ai programmatori di computer.

Ebbene con questo programma potrete effettuare le vostre proprie forme basate sugli stessi principi usati da Escher.

Le figure che avrete a disposizione sembreranno un mosaico complicato, ma se presterete

particolare attenzione, vi accorgete che si tratta della stessa forma posizionata ogni volta in modo differente. Escher ha appunto studiato questo tipo di forme ed ha inventato un gioco basato su di esse.

Inizialmente vi verrà chiesto il formato della tavola di base, dopodichè avrà inizio il gioco vero e proprio: dando ogni volta la direzione del quadrato, il vostro TI vi mostrerà la vostra indubitabile fantastica creazione, come se si trattasse di un'opera originale di Escher.

Buon divertimento!

P.S. Per partire con altri grafici premete -R-).

```

100 REM *****
110 REM ** ESCHER **
120 REM *****
130 REM TI-99 BASIC
140 CALL CLEAR
150 DATA 0000000000000000,000000
0000000000,0000000000000000,0000
000000000000

```

```

160 DATA 8080808080808080,FF0000
0000000000,0101010101010101,0000
0000000000FF
170 DATA 2020202020202020,0000FF
0000000000,0404040404040404,0000
000000FF0000
180 DATA 1010101010101010,000000
FF00000000,8080808080808080,0000

```



```

0000FF000000
190 DATA C0C0C0C0C0C0C0C0,FFFF00
0000000000,0303030303030303,0000
00000000FFFF
200 DATA F0F0F0F0F0F0F0F0,FFFFFF
FF00000000,0F0F0F0F0F0F0F0F,0000
0000FFFFFFF
210 DATA 8040201008040201,010204
0810204080,8040201008040201,0102
040810204080
220 DATA 101020C000000000,808040
3000000000,0000000030408080,0000
0000C0201010
230 DATA FFFFCF8F0E0C080,FF7F3F
1F0F070301,0103070F1F3F7FFF,80C0
E0F0F8FCFEFF
240 DATA F0F0F0F000000000,0F0F0F
0F00000000,000000000F0F0F0F,0000
0000F0F0F0F0
250 DATA 00183C7E7E3C1800,00183C
7E7E3C1800,00183C7E7E3C1800,0018
3C7E7E3C1800
260 DATA 8142241818244281,814224
1818244281,8142241818244281,8142
241818244281
270 DATA 08080808FF080808,101010
10FF101010,101010FF10101010,0808
08FF08080808
280 DATA AA55AA55AA55AA55,55AA55
AA55AA55AA,AA55AA55AA55AA55,55AA
55AA55AA55AA
290 DATA F0F0F0F00F0F0F0F,0F0F0F
0FF0F0F0F0,F0F0F0F00F0F0F0F,0F0F
0F0FF0F0F0F0
300 FOR TEL=96 TO 155
310 READ X$
320 CALL CHAR(TEL,X$)
330 NEXT TEL
340 PRINT "DISPONI DEI SEGUENTI
CARATTERI:"
350 FOR TEL=96 TO 155 STEP 4
360 PRINT TEL;
370 CALL HCHAR(24,10,TEL)
380 PRINT
390 NEXT TEL
400 INPUT "FORMATO TAVOLA BASE(3
-8)":GR
410 IF (GR<3)+(GR>8)<>0 THEN 400
420 HH=INT(24/GR)
430 HHH=HH*HH
440 DIM D(64)
450 DIM RI(64)
460 II=0
470 FOR CC=1 TO GR*GR
480 PRINT "CARATTERE NR: ";CC;
490 INPUT D(CC)
500 IF (D(CC)<96)+(D(CC)>155)+(D
(CC)/4-INT(D(CC))/4<>0)<>0 THEN
480
510 NEXT CC
520 FOR CC=1 TO HHH

```

```

530 PRINT "DIREZIONE TAVOLA ";CC
;"00-3): ";
540 INPUT RI(CC)
550 IF (RI(CC)<0)+(RI(CC)>3)<>0
THEN 530
560 NEXT CC
570 CALL CLEAR
580 CALL SCREEN(10)
590 AAA=0
600 FOR OB1=1 TO HH*GR STEP GR
610 FOR OB2=4 TO HH*GR STEP GR
620 AAA=AAA+1
630 I=0
640 ON RI(AAA)+1 GOSUB 680,740,8
00,860
650 NEXT OB2
660 NEXT OB1
670 GOTO 950
680 FOR C1=OB1 TO OB1+GR-1
690 FOR C2=OB2 TO OB2+GR-1
700 GOSUB 920
710 NEXT C2
720 NEXT C1
730 RETURN
740 FOR C2=OB2+GR-1 TO OB2 STEP
-1
750 FOR C1=OB1 TO OB1+GR-1
760 GOSUB 920
770 NEXT C1
780 NEXT C2
790 RETURN
800 FOR C1=OB1+GR-1 TO OB1 STEP
-1
810 FOR C2=OB2+GR-1 TO OB2 STEP
-1
820 GOSUB 920
830 NEXT C2

```

Guida per l'input dei programmi sul TI 99/4A

A causa del sistema di codifica utilizzato, possono capitare in alcune linee delle improvvise "andate a capo" delle quali, naturalmente, non va tenuto conto in fase di digitazione.

Vi potrà capitare, a volte di trovare nei listati di programmi per TI 99/4a alcuni caratteri sottolineati. La sottolineatura rappresenta una particolare codifica dei tasti di controllo: dovrete quindi premere il tasto indicato dal carattere insieme al tasto CONTROL.

Ad esempio, se trovate A dovrete battere CONTROL+A; se trovate —, dovrete battere CONTROL+SPAZIO, e così via.


```

840 NEXT C1
850 RETURN
860 FOR C2=OB2 TO OB2+GR-1
870 FOR C1=OB1+GR-1 TO OB1 STEP
-1
880 GOSUB 920
890 NEXT C1
900 NEXT C2
910 RETURN

```

```

920 I=I+1
930 CALL HCHAR(C1,C2,D(I)+RI(AAA
))
940 RETURN
950 CALL KEY(0,K,S)
960 MSG$="ALTRI GRAFICI?(S/N) "
970 IF S=0 THEN 32767
980 IF K=82 THEN 150
990 END

```



16/48K

Ring

Credete di essere in grado di affrontare in un serrato corpo a corpo... il vostro Spectrum? Attenti a non sottovalutare il vostro computer solo perché non è dotato di mani e piedi: non solo lo Spectrum è in grado di gestire efficientemente e con agilità gli arti che questo programma gli fornisce, ma soprattutto ha un caratterino permaloso che lo rende avversario temibile e instancabile. Se, nonostante ciò, vi sentite all'altezza del vostro avversario, battete 'Ring' e salvateo con GO TO 9000. Una volta lanciato il programma, caratterizzato da grafica estesa e divertente, vi troverete su una delle più prestigiose pedane del mondo, circondati da un immenso numero di tifosi, giunti da ogni dove per vedere chi sarà il nuovo campione. Garantisce la regolarità dell'incontro un robot, che ricopre per l'occasione il ruolo di arbitro. E ve-

niamo al combattimento vero e proprio: in realtà non si tratta di boxe in senso stretto, bensì di qualcosa più vicino al Savate. Infatti durante il combattimento potete sferrare (e ricevere) pugni e calci. I tasti da utilizzare per controllare il vostro rappresentate sono '8' & '9', per muovervi a sinistra-destra; '5' per parare i colpi dell'avversario; '6' per far partire il vostro micidiale destro; infine '7' per sferrare pedate. Una precisazione va fatta per quanto concerne il controllo tramite tastiera: dato che il computer deve controllare il vostro personaggio, l'arbitro ed il proprio pugile, i tasti vanno premuti a lungo in modo da dare al computer occasione di leggerli non appena ha ultimato altre operazioni. Vince il match chi riesce ad atterrare nove volte il proprio avversario. Buon combattimento.

```

1 REM *****
2 REM ***** R I N G *****
3 REM *****
5 BORDER 1: CLS
10 REM *****
11 REM ****VARIABILI & UDG****
12 REM *****

```

```

15 FOR i=0 TO 159: READ x:
POKE USR "a"+i,x: NEXT i
20 DATA 224,240,224,192,192,22
4,224,224
21 DATA 240,193,255,128,128,12
8,128,192
22 DATA 7,15,7,3,3,7,7,7
23 DATA 15,131,255,1,1,1,1,3
24 DATA 224,240,224,192,193,25
5,193,192
25 DATA 192,192,192,160,144,13
6,136,204

```



```

26 DATA 230,244,228,196,196,25
   2,192,192
27 DATA 103,47,39,35,35,63,3,3

28 DATA 3,3,5,9,17,17,17,51
29 DATA 7,15,7,3,131,255,131,3

30 DATA 56,56,56,16,16,124,84,
   84
31 DATA 214,16,16,40,68,68,68,
   198
32 DATA 0,0,0,0,64,224,255,255

33 DATA 0,0,0,0,0,0,129,255
34 DATA 0,0,0,0,2,7,255,255
35 DATA 126,66,66,165,129,189,
   129,126
36 DATA 7,6,7,7,7,63,47,47
37 DATA 47,39,35,114,2,2,2,14
38 DATA 224,96,224,224,224,252
   ,244,244
39 DATA 244,228,196,78,64,64,6
   4,112
40 GO SUB 5000
45 LET es=0: LET ms=0:
   BORDER 1: PAPER 7
50 LET m=17: LET e=14
60 LET goto=0

3000 REM *****
3001 REM ****MOSSA GIOCATORE****
3002 REM *****

3010 LET a$="K": LET b$="L":
   LET c$="K": LET d$="L"
3020 PRINT AT 9,m; INK 1;c$
3030 PRINT AT 10,m; INK 1;d$
3100 IF RND>.5 THEN LET goto=1:
   GO TO 3500
3140 IF INKEY$="" THEN GO TO 35
   00
3150 IF INKEY$<"5" OR INKEY$>"9"
   THEN GO TO 3500
3200 IF INKEY$<"8" THEN GO TO 3
   300
3210 PRINT AT 9,m;" "
3220 PRINT AT 10,m;" "
3230 LET m=m+(INKEY$="9" AND m<1
   9)-(INKEY$="8")
3270 GO TO 3330
3300 IF INKEY$="5" THEN LET c$=
   "H": LET d$="I": BEEP .09,6
3310 IF INKEY$="6" THEN LET c$=
   "J": LET d$="I": BEEP .09,8
3320 IF INKEY$="7" THEN LET c$=
   "C": LET d$="D": BEEP .09,1
   0

```

```

3330 PRINT AT 9,m; INK 1;c$
3340 PRINT AT 10,m; INK 1;d$
3360 IF goto=1 THEN LET goto=0:
   GO TO 3600

3407 REM *****
3408 REM ****MOSSA COMPUTER****
3409 REM *****

3500 IF RND<.5 THEN GO TO 3600
3520 LET z=INT (RND*9)+1
3530 IF z<5 THEN LET a$="A":
   LET b$="B": BEEP .1,10
3540 IF z=5 THEN LET a$="G":
   LET b$="F": BEEP .1,6
3550 IF z=6 THEN PRINT AT 9,e;"
   ";AT 10,e;" ": LET e=e+1
3560 IF z=7 THEN PRINT AT 9,e;"
   ";AT 10,e;" ": LET e=e-1:
   IF e=12 THEN LET e=13
3580 IF z=9 THEN LET a$="K":
   LET b$="L"
3590 PRINT AT 9,e; INK 2;a$
3570 IF z=8 THEN LET a$="E":
   LET b$="F": BEEP .1,8

3591 PRINT AT 10,e; INK 2;b$
3592 IF goto=1 THEN GO TO 3100

3597 REM *****
3598 REM ****URTO FRONTALE****
3599 REM *****

3600 IF e>=m THEN PRINT AT 9,12
   ;" ";AT 10,12; INK 2
   ;" MN"; INK 1;" NO":
   BEEP .5,5: BEEP .5,5:
   BEEP .09,6: BEEP .09,7:
   BEEP .9,6: BEEP .09,5:
   PRINT AT 10,13;" "
   GO TO 50

3617 REM *****
3618 REM ****ARBITRO****
3619 REM *****

3620 IF RND<.5 THEN GO TO 3650
3630 PRINT AT 6,13; INK 4;" QS"
   ;AT 7,13; INK 4;" RT"
3640 GO TO 4000
3650 PRINT AT 6,13; INK 4;"QS "
   ;AT 7,13; INK 4;"RT "

```



```

4000 REM *****
4001 REM *****PUNTI*****
4002 REM *****
4010 IF m<>e+1 OR a$="G"
      THEN GO TO 3000
4030 IF a$="K" THEN GO TO 4100
4040 IF c$<>"K" THEN GO TO 3000
4050 LET es=es+1
4060 PRINT AT 9,e-1; INK 2;"K "
      ;AT 10,e-1; INK 2;"L";
      INK 1;"NO"
4070 BEEP .3,5: BEEP .3,5:
      BEEP .09,5: BEEP .09,6:
      BEEP .09,5: BEEP .09,4
4080 PRINT AT 9,e-1;" ";AT 10,e-
      1;" "
4090 GO TO 4200
4100 IF c$<>"J" AND d$<>"D"
      THEN GO TO 3000
4110 LET ms=ms+1
4120 PRINT AT 9,e; INK 1;" K";
      AT 10,e-1; INK 2;"MN";
      INK 1;"L"
4130 BEEP .3,10: BEEP .3,10:
      BEEP .09,11: BEEP .09,12:
      BEEP .09,10: BEEP .09,13
4140 PRINT AT 9,e+1;" ";AT 10,e-
      1;" "
4200 PRINT AT 1,8;"{SG8}PUNTI=";
      es;AT 1,17;"PUNTI=";ms;"
      {SG8}"
4500 REM *****
4501 REM ***VITTORIA/SCONFITTA**
4502 REM *****
4510 IF ms<9 THEN GO TO 4550
4515 CLS
4520 FOR i=0 TO 7: FOR o=7 TO 0
      STEP -1
4525 BORDER o
4530 PRINT AT 1,0; INK i;"CONGRA
      TULAZIONI! HAI": GO SUB 60
      00

```

```

4535 PRINT AT 16,10; INK i;"PER
      ";ms;" A ";es
4540 BEEP o/1000,i: NEXT o:
      NEXT i: BEEP .09,15:
      BEEP .09,15: BEEP .1,18:
      CLS : GO TO 40
4550 IF es<9 THEN GO TO 50
4560 CLS : PRINT AT 1,0;"LO SPEC
      TRUM HA": PAUSE 75: LET i=2
      : GO SUB 6000
4565 PRINT AT 16,10;"PER ";es;"
      A ";ms
4570 BEEP .7,7: BEEP .7,8:
      BEEP .7,7: BEEP .6,6:
      BEEP .6,7: BEEP .6,3:
      BEEP 1.5,-2: CLS : GO TO 4
      0
4997 REM *****
4998 REM *ALLESTIMENTO DISPLAY*
4999 REM *****
5000 FOR i=0 TO 16: FOR o=0 TO 9
      : PRINT AT i,o; INK 0;
      PAPER INT (RND*5)+2;"P";
      AT i,o+22;"P": NEXT o:
      NEXT i
5001 FOR i=15 TO 16: FOR o=10
      TO 21: PRINT AT i,o; INK 0
      ; PAPER INT (RND*5)+2;"P":
      NEXT o: NEXT i
5010 PRINT AT 1,8;"{SG8}PUNTI=0
      PUNTI=0{SG8}"
5020 FOR i=8 TO 24: PRINT AT 0,i
      ;"{SG8}";AT 2,i;"{SG8}":
      NEXT i
5030 PLOT 128,140: DRAW 37,0:
      DRAW 0,-75: DRAW -75,0:
      DRAW 0,75: DRAW 38,0
5040 PRINT AT 17,0;"{33SG8}5=PAR
      ATA{3SG8}6=PUGNO{3SG8}7=CAL
      CIO{38SG8}8=SINISTRA{4SG8}9
      =DESTRA{38SG8}"
5050 RETURN
6000 PRINT AT 7,0; INK i;"
      {SG8} {SG8} {SG8} {SG8}
      {SG8} {5SG8} {3SG8}
      {SG8} {SG8} {SG8}
      {2SG8} {SG8} {SG8}
      {SG8} {SG8} {SG8}
      {SG8} {SG8} {SG8} {SG8}
      {SG8} {SG8} {SG8}
      {SG8} {SG8} {SG8}

```



```

{SG8} {SG8} {2SG8}
{SG8} {SG8} {SG8}
{SG8} {SG8} {SG8} {SG8}
{SG8} {SG8} {SG8}
{3G8}{SG8} {SG8}
{SG8} {SG8} {SG8}
{SG8} {3SG8}"

```

```

6010 RETURN
9000 REM *****

```

```

9001 REM ****SALVA & VERIFICA***
9002 REM *****
9010 CLS : SAVE "RING" LINE 5
9020 PRINT AT 10,0;"RIAVVOLGI IL
      NASTRO E CONNETTI EAR PER
      VERIFICARE IL PROGRAMMA."
9030 VERIFY "RING"
9040 CLS : PRINT AT 13,14;
      FLASH 1;"O.K."

```

Scappa...



16/48K

Vi siete introdotti furtivamente in un grande labirinto alla ricerca dei favolosi tesori (gemme e spade) che intere generazioni di gnomi e folletti, distratti e disordinati, hanno lanciato in giro qua e là. In realtà in questo gioco l'ostacolo maggiore è rappresentato, più che dal labirinto (semplice e completamente in vista), dalla presenza di una gigantesca palla rimbalzante, posta ad eterna custodia dei tesori del palazzo. La palla non smette mai di inseguirvi, ma può in taluni casi imprigionarsi in un vicolo cieco nel tentativo di giungere nei vostri pressi. La corretta strategia di gioco richiede proprio che voi vi muoviate in modo da disorientare la palla: una volta che essa si è imprigionata da sola, potrete muovervi con maggior tranquillità fra i muri del labirinto.

I premi da raccogliere aumentano man mano di numero; quando avrete raccolto spade e gioielli presenti in un dato percorso dovrete uscire dai passaggi segreti (non tanto segreti) per conquistare il bonus ed il diritto a passare allo stadio successivo. Per inciso, tali passaggi costituiscono importante strumento per confondere la palla magica. Il programma utilizza un gran numero di caratteri grafici (da segnalare il simbolo che rappresenta il giocatore) e, una volta lanciato, vi consente di scegliere fra il labirinto visibile o invisibile: quest'ultimo dovrebbe essere selezionato solo dopo aver acquistato notevole familiarità con il percorso standard. Il programma può essere salvato con GO TO 9990.

```

5 GO SUB 9000
10 OVER 0: BORDER 0: PAPER 0:
  INK 9: CLS
15 GO SUB 8000
20 DIM a$(10,64)
21 LET sc=0: LET li=3
24 LET items=3
25 LET t=0: LET x=17: LET y=2:
  LET ax=3: LET ay=28

```

```

26 LET items=items+1
30 RESTORE 9100: FOR a=1 TO 10
  : READ a$(a): NEXT a
40 GO SUB 600+gco: REM stampa
  il labirinto
100 GO SUB 500: REM stampa uomo
105 GO SUB 550: REM stampa palla
  a rimbalzante
110 IF INKEY$="8" THEN

```



```

GO SUB 500: LET y=y+(2
AND a$( ((x+1)/2),y+3)<>"
A"): BEEP .001,y+20:
GO SUB 500
115 IF INKEY$="5" THEN
GO SUB 500: LET y=y-(2
AND a$( ((x+1)/2),y)<>"B"
): BEEP .001,50-y: GO SUB 5
00
120 IF INKEY$="6" THEN
GO SUB 500: LET x=x+(2
AND a$( (x/2)+1,y+1)<>"A"
): BEEP .001,40-y: GO SUB 5
00
125 IF INKEY$="7" THEN
GO SUB 500: LET x=x-(2
AND a$( ((x+1)/2)-1,y+1)
<>"A"): BEEP .001,45-y:
GO SUB 500
130 IF a$( ((x+1)/2),y+1)="E"
THEN LET t=t+1: LET sc=sc
+100: FOR z=10 TO 40 STEP 5
: BEEP .005,z: NEXT z:
LET a$( (x+1)/2,y+1)=" ":
LET a$( ((x+1)/2)+1,y+33)="
": PRINT OVER 1;AT x,y;"E
";AT x+1,y;"F"
135 IF a$( ((x+1)/2),y+1)="G"
THEN LET t=t+1: LET sc=sc
+200: FOR z=20 TO 60 STEP 5
: BEEP .005,z: NEXT z:
LET a$( (x+1)/2,y+1)=" ":
LET a$( ((x+1)/2)+1,y+33)="
": PRINT OVER 1;AT x,y;"G
H";AT x+1,y;"IJ"
140 IF t>=items AND (x<=2 OR x
>=19) AND y=14 THEN
GO TO 700
145 IF x>=19 AND y=14 THEN
GO SUB 500: LET x=3: LET y
=14: GO SUB 500
150 IF x<=2 AND y=14 THEN
GO SUB 500: LET x=17:
LET y=14: GO SUB 500
155 PRINT AT 21,7; PAPER 2;
BRIGHT 1;sc; BRIGHT 0;" ";
BRIGHT 1;AT 21,27;li;
BRIGHT 0;" "
160 IF x=ax AND y=ay THEN
GO SUB 800
161 LET o=ax: LET p=ay
165 IF ay>y THEN GO SUB 550:
LET ay=ay-(2 AND a$( (ax+1)
/2,ay)<>"B"): GO SUB 550
170 IF ay<y THEN GO SUB 550:
LET ay=ay+(2 AND a$( (ax+1)

```

```

/2,ay+3)<>"A"): GO SUB 55
0
175 IF ax<19 THEN IF ax>x
THEN GO SUB 550: LET ax=a
x-(2 AND a$( ((ax+1)/2)-1,ay
+1)<>"A"): GO SUB 550
180 IF ax>2 THEN IF ax<x
THEN GO SUB 550: LET ax=a
x+(2 AND a$( ((ax+1)/2)+1,ay
+1)<>"A"): GO SUB 550
190 IF ax=o AND ay=p THEN
GO SUB 550: LET ay=ay+(2
AND a$( ((ax+1)/2),ay+3)
<>"A")-(2 AND a$( ((ax+1)/
2),ay)<>"B"): GO SUB 550
200 GO TO 110
500 PRINT AT x,y; OVER 1;
BRIGHT 1; INK 6;"KL"
505 PRINT AT x+1,y; OVER 1;
BRIGHT 1; INK 6;"MN"
510 RETURN
550 PRINT AT ax,ay; OVER 1;
BRIGHT 1; INK 4;"OP"
555 PRINT AT ax+1,ay; OVER 1;
BRIGHT 1; INK 4;"QR"
560 RETURN
600 FOR a=1 TO 10: BEEP .001,40
+a: PRINT AT (a*2)-1,0;
INK 4;a$(a): NEXT a
601 PRINT AT 21,0; PAPER 2;"PUN
TI ";sc;" ";AT 21,22;"VITE
";li;" "
605 LET z$="E F ": FOR z=1 TO i
tems
610 LET rx=(INT (RND*8)+1)*2:
LET ry=(INT (RND*14)+1)*2
615 IF a$( (rx+2)/2,ry+1)<>" "
THEN GO TO 610
620 PRINT INK INT (RND*6)+2;
AT rx+1,ry;z$( TO 2);AT rx+
2,ry;z$(3 TO )
621 LET a$( (rx+2)/2,ry+1)=z$(1)
: LET a$( (rx+2)/2,ry+2)=z$(
2)
622 LET a$( (rx+2)/2,ry+3)=z$(3
): LET a$( (rx+2)/2,ry+32)=z
$(4)
625 LET z$=("GHIJ" AND z>=2)+("
E F " AND z<=1): NEXT z
630 RETURN
700 FOR a=1 TO 50
701 OUT 254,7
702 OUT 254,130
703 OUT 254,30
704 OUT 254,16
705 OUT 254,255

```



```

706 OUT 254,201
707 OUT 254,150
708 NEXT a
710 BEEP .002,a
715 FOR a=0 TO 10: FOR b=1 TO 7
  : BEEP .02,(a+b)+RND*10:
  PRINT AT 10,10; PAPER b;
  INK 9; FLASH 1;" B O N U S
  ": NEXT b: NEXT a

720 CLS : LET sc=sc+500
730 GO TO 25
800 FOR b=1 TO 5: FOR a=0 TO 7:
  BORDER a: BEEP .005,a*b:
  NEXT a: NEXT b: BORDER 0

805 LET li=li-1: PRINT AT 21,27
  ; PAPER 2; BRIGHT 1;li;
  BRIGHT 0;" "

810 PRINT AT 10,7; FLASH 1;
  PAPER 2;"HAI PERSO UNA VIT
  A"
815 FOR a=1 TO 100: PAUSE 1:
  NEXT a

816 IF li<=0 THEN GO TO 830
820 CLS : GO TO 25
830 PRINT INK 9;AT 10,2;
  PAPER 3; FLASH 1;" LA P
  ARTITA E' FINITA "

840 PAUSE 100: PRINT AT 13,2;
  INVERSE 1;" Un'altra part
  ita (S/N) ? ": PAUSE 0
850 IF INKEY$="s" OR INKEY$="S"
  THEN RUN
855 IF INKEY$="n" OR INKEY$="N"
  THEN STOP
860 GO TO 850
8000 REM istruzioni

8005 PRINT AT 0,0; PAPER 2;
  INK 9; BRIGHT 1;" SC
  APPA VELOCE.... "

8020 PRINT AT 2,0;"...dal grand
  e labirinto ove ti sei intr
  odotto alla ricerca di pre
  ziose gemme e magiche spade
  , prima che la misteriosa p
  alla saltellante ti schia
  cci sotto il suo formidabile
  peso!"
8025 PRINT "Quando avrai raccolt
  o tutti i premi,sfrutta i
  passaggi segreti per guadag
  nare il BONUS e per passa
  re allo stadio successivo."
8030 PRINT " E --> SPADA K
  L --> TU F
  MN GH -->
  GEMMA OP --> PALLA IJ
  QR"

```

```

8035 PRINT AT 18,0;"TASTI: 5,6,7
  & 8"
8100 FOR a=0 TO 7: BEEP .002,(
  RND*10)+a
8105 PRINT AT 20,0; INVERSE 1;"P
  REMI==> 1 LABIRINTO VISIBI
  LE ==> 2 LABIRINTO I
  NVISIBLE"
8110 LET u$=INKEY$
8115 IF u$="1" OR u$="2" THEN
  LET gco=VAL u$-1: CLS :
  RETURN
8120 NEXT a: GO TO 8100

9000 REM udg
9005 RESTORE 9000: FOR n=USR "a"
  TO USR "r"+7: READ a:
  POKE n,a: NEXT n
9006 RETURN
9010 DATA 255,128,192,224,255,24
  0,240,240
9020 DATA 254,2,6,14,254,30,30,3
  0
9030 DATA 240,240,240,255,224,19
  2,128,255
9040 DATA 30,30,30,254,14,6,2,25
  4
9045 DATA 0,0,0,8,8,28,28,28
9050 DATA 28,28,62,29,30,30,0,0
9055 DATA 0,0,0,15,16,32,64,63
9060 DATA 0,0,0,240,16,8,4,252
9065 DATA 64,64,32,16,8,4,2,1
9070 DATA 2,4,8,16,32,64,128,0
9075 DATA 63,63,32,255,127,99,99
  ,103
9080 DATA 248,248,8,254,248,16,2
  52,252
9085 DATA 79,79,64,63,7,31,24,16
9090 DATA 120,56,128,240,224,224
  ,24,12
9095 DATA 0,0,0,15,31,63,63,63
9096 DATA 0,0,0,240,248,252,252,
  252
9097 DATA 63,63,63,63,31,15,0,0
9098 DATA 252,252,252,252,248,24
  0,0,0
9100 REM dati labirinto
9105 DATA "ABABABABABABAB ABABA
  BABABABABABCDCDCDCDCDCDCDCDC
  CDCDCDCDCDCDCDCDCDCDCDCDCDCDCDC"
9110 DATA "AB
  ABCD
  CD"
9115 DATA "AB ABABABABABABABAB
  ABABAB ABCD CDCDCDCDCDCDCDCDC
  CDCDCDCDCDCDCDCDCDCDCDCDCDCDCDC"

```



```

9120 DATA "AB AB AB AB
           AB AB ABCD CD CD
           CD CD CD CD"
9125 DATA "AB AB AB AB AB AB
           AB ABCD CD CD CDCD
           CD CD CD"
9130 DATA "AB
           AB ABCD AB CD
           CD CD"
9135 DATA "AB AB ABAB AB A
           BAB ABCD CD CDCD
           CD CDCD CD"
9140 DATA "AB AB AB AB A
           BAB ABABCD CD CD
           CD CDCD CDCD"
9145 DATA "AB
           ABCD CD
           CD"

```

```

9150 DATA "ABABABABABAB ABABA
           BABABABABABCD CDCDCDCDCDCDC
           CDCDCDCDCDCDCDCDCDCDC"
9800 STOP
9990 SAVE "scappa.." LINE 9995
9992 PRINT AT 10,0;"Riavvolgi il
           nastro & connetti EAR per
           verificare il programma."
9993 VERIFY "scappa..": CLS :
           PRINT AT 10,14;"O.K."
9994 STOP
9995 CLS : PRINT AT 10,8;
           FLASH 1; PAPER 5; INK 9;"F
           ERMA IL NASTRO"
9998 FOR a=1 TO 3: FOR b=1 TO 20
           : BEEP .02,(a+b)*2: NEXT b:
           NEXT a: RUN

```

Database



Stampante e disk drive opzionale

Questo programma può servirvi a memorizzare i dati relativi alle persone che conoscete; i dati comprendono nome, indirizzo, città, provincia, codice postale e telefono (con prefisso). Le informazioni possono poi essere visualizzate sullo schermo o stampate su carta, creando in tal modo una rubrica. I dati vengono automaticamente salvati su nastro o su disco, e richiamati ogni volta che fate girare il programma.

"Database" è molto facile da usare. Il suo menù comprende 4 opzioni:

```

100 MEM=200: LW=1: POKE53280,0: POK
    E53281,0
110 PRINT"{CLR}{ 3 GIU' }{YEL}
        { 9 SPAZI }{RVS}C{OFF}ASSETTA
        OPPURE {RVS}D{OFF}ISCO?"
120 GETDV$: IFDV$<>"C"ANDDV$<>"D"
    THEN120
130 IFDV$="D"THEN170
140 PRINT"{ 3 GIU' }{ 4 SPAZI }RUB
        RICA GIA' REGISTRATA
        { 2 SPAZI }(S/N)?"

```

* *aggiunta* - serve ad aggiungere nuovi nominativi alla rubrica. Se commettete un errore di battitura, potrete correggere il dato sbagliato dopo aver digitato tutte le informazioni.

* *Cancellazione* - serve a cancellare dalla rubrica un singolo nominativo.

* *Stampa* - può avvenire sia su video che tramite la stampante.

* *Fine* - da selezionare al termine delle operazioni. Provoca il salvataggio delle informazioni presenti in memoria e il reset del C64.

```

150 GETG$: IFG$<>"S"ANDG$<>"N"THE
    N150
160 IFG$="N"THENLW=0
170 DIMQ(MEM), A$(MEM,6), CT(MEM),
    CY(6)
180 GOTO380
190 FORJ=1TO2500:NEXT
200 RETURN
210 POKE53265, PEEK(53265)AND239
220 FORJ=1TO10:NEXT
230 RETURN

```



```

240 POKE53265,PEEK(53265)OR16
250 FORJ=1TO10:NEXT
260 RETURN
270 S1=54272
280 FORL=0TO24:POKES1+L,0:NEXT
290 POKES1+14,5:POKES1+18,16:POK
ES1+3,1:POKES1+24,143
300 POKES1+6,240:POKES1+4,65:FR=
5389
310 FORT=1TO5
320 FQ=FR+PEEK(S1+27)*3.5
330 HF=INT(FQ/256):LF=FQ-HF*256
340 POKES1,LF:POKES1+1,HF
350 NEXT
360 POKES1+24,0
370 RETURN
380 P$="{HOME}{ 26 GIU' }":T4=28:
GOSUB210
390 POKE53280,7:POKE53281,1:T1=3
:T2=10:T3=15:T5=13
400 PRINT"{CLR}{RVS}{<4>}
{ 7 SPAZI}{BLU}U{ 24 C}I
<4>}{ 6 SPAZI}";
410 PRINT"{ 8 SPAZI}{BLU}B{WHT}
{ 24 SPAZI}{BLU}B<4>
{ 6 SPAZI}";
420 PRINT"{ 8 SPAZI}{BLU}J
{ 24 C}K<4>}{ 7 SPAZI}"
430 PRINT"{HOME}{GIU' }"SPC(11)"
{RED} RUBRICA INDIRIZZI{BLK}
"
440 PRINT"{GIU'}{RVS}{<7>}
{ 2 SPAZI}U{ 4 C}I
{ 2 SPAZI}U{ 25 C}I
{ 3 SPAZI}";
450 PRINT"{BLK}1<7> B{OFF}
{ 4 SPAZI}{RVS}B{ 2 SPAZI}B
{OFF}{ 25 SPAZI}{RVS}B
{ 2 SPAZI}";
460 PRINT"{ 3 SPAZI}J{ 4 C}K
{ 2 SPAZI}J{ 25 C}K
{ 3 SPAZI}";
470 PRINT"{ 2 SPAZI}U{ 9 C}I
{ 2 SPAZI}U{ 18 C}I
{ 5 SPAZI}";
480 PRINT"{BLK}2<7> B{OFF}
{ 9 SPAZI}{RVS}B{ 2 SPAZI}B
{OFF}{ 18 SPAZI}{RVS}B
{ 4 SPAZI}";
490 PRINT"{ 3 SPAZI}J{ 9 C}K
{ 2 SPAZI}J{ 18 C}K
{ 5 SPAZI}";
500 PRINT"{ 2 SPAZI}U{ 6 C}I
{ 5 SPAZI}U{ 12 C}I
{ 11 SPAZI}";
510 PRINT"{BLK}3<7> B{OFF}
{ 6 SPAZI}{RVS}B{ 5 SPAZI}B
{OFF}{ 12 SPAZI}{RVS}B
{ 10 SPAZI}";
520 PRINT"{ 3 SPAZI}J{ 6 C}K
{ 5 SPAZI}J{ 12 C}K
{ 11 SPAZI}";

```

```

530 PRINT"{ 2 SPAZI}U{ 5 C}I
{ 4 SPAZI}U{ 2 C}I
{ 4 SPAZI}U{ 3 C}I
{ 2 SPAZI}U{ 5 C}I
{ 5 SPAZI}";
540 PRINT"{BLK}4<7> B{OFF}
{ 5 SPAZI}{RVS}B{ 4 SPAZI}B
{OFF}{ 2 SPAZI}{RVS}B
{ 2 SPAZI}{BLK}5<7> B{OFF}
{ 3 SPAZI}{RVS}B{ 2 SPAZI}B
{OFF}{ 5 SPAZI}{RVS}B
{ 4 SPAZI}";
550 PRINT"{ 3 SPAZI}J{ 5 C}K
{ 4 SPAZI}J{ 2 C}K
{ 4 SPAZI}J{ 3 C}K
{ 2 SPAZI}J{ 5 C}K
{ 5 SPAZI}";
560 PRINT"{ 2 SPAZI}U{ 4 C}I
{ 5 SPAZI}U{ 13 C}I
{ 12 SPAZI}";
570 PRINT"{BLK}6<7> B{OFF}
{ 4 SPAZI}{RVS}B{ 5 SPAZI}B
{OFF}{ 4 SPAZI}{RVS}{BLK}/
{OFF}{ 8 SPAZI}<7>{RVS}B
{ 11 SPAZI}";
580 PRINT"{ 3 SPAZI}J{ 4 C}K
{ 5 SPAZI}J{ 13 C}K
{ 12 SPAZI}";
590 PRINT"{ 40 SPAZI}";
600 PRINT"{ 40 SPAZI}"
610 PRINTLEFT$(P$,5)SPC(T1)"
{BLK}{RVS}NOME"
620 PRINTLEFT$(P$,8)SPC(T1)"
{RVS}INDIRIZZO"
630 PRINTLEFT$(P$,11)SPC(T1)"
{RVS}CITTA'"
640 PRINTLEFT$(P$,14)SPC(T1)"
{RVS}PROV."SPC(T1+11)"CAP"
650 PRINTLEFT$(P$,17)SPC(T1)"
{RVS}TEL."
660 PRINT"{ 3 GIU'}{ 2 SPAZI}
{RVS}{RED} A {OFF}{BLK} AGGI
UNGI"SPC(8)"{RVS}{RED} C
{OFF}{BLK} CANCELLA"
670 PRINT"{GIU'}{ 2 SPAZI}{RVS}
{RED} S {OFF}{BLK} STAMPA"SP
C(10)"{RVS} F {OFF} FINE"
680 FORJ=1TO100:NEXT:GOSUB270:GO
SUB240
690 FORJ=1TO2000:NEXT
700 IFLW=1THENLW=0:GOTO2000
710 O=O+1:PRINTLEFT$(P$,2)SPC(11
)"<4>FAI LA TUA SCELTA ":I
FO=3THENO=0:GOTO750
720 FORJ=1TO500:NEXT:PRINTLEFT$(
P$,2)SPC(11)"{ 18 SPAZI}"
730 FORJ=1TO100:NEXT
740 GOTO710
750 GETG$:IFG$=""THEN750
760 SS$="":SL$=""
770 IFG$="A"THEN820
780 IFG$="C"THEN2290
790 IFG$="S"THEN2510

```



```

800 IFG$="F" THEN 2940
810 GOTO 750
820 Q=Q+1
830 FORJ=1 TO 100: NEXT: GOSUB 270
840 PRINTLEFT$(P$, 2) SPC(11) "
    [<4>]{ 6 SPAZI}NOME?
    { 7 SPAZI}{BLK}"
850 FORX=1 TO 25
860 PRINTLEFT$(P$, 5) SPC(T2+X) "
    [<+>]"
870 FORJ=1 TO 5: NEXT
880 PRINTLEFT$(P$, 5) SPC(T2+X) " "
    ;
890 GETG$: GOSUB 3030: IFG$="" THEN 860
900 IFG$=CHR$(13) THEN POKE 198, 0: GOTO 940
910 PRINTLEFT$(P$, 5) SPC(T2+X) G$
920 A$(Q, 1)=A$(Q, 1)+G$
930 NEXT
940 IFLC=1 THEN LC=0: GOTO 1630
950 FORJ=1 TO 100: NEXT: GOSUB 270
960 PRINTLEFT$(P$, 2) SPC(11) "
    [<4>]{ 4 SPAZI}INDIRIZZO?
    { 4 SPAZI}{BLK}"
970 FORX=1 TO 18
980 PRINTLEFT$(P$, 8) SPC(T3+X) "
    [<+>]"
990 FORJ=1 TO 5: NEXT
1000 PRINTLEFT$(P$, 8) SPC(T3+X) "
    ";
1010 GETG$: GOSUB 3030: IFG$="" THEN 980
1020 IFG$=CHR$(13) THEN POKE 198, 0: GOTO 1060
1030 PRINTLEFT$(P$, 8) SPC(T3+X) G$
1040 A$(Q, 2)=A$(Q, 2)+G$
1050 NEXT
1060 IFLC=1 THEN LC=0: GOTO 1630
1070 FORJ=1 TO 100: NEXT: GOSUB 270
1080 PRINTLEFT$(P$, 2) SPC(11) "
    [<4>]{ 5 SPAZI}CITTA'?
    { 6 SPAZI}{BLK}"
1090 FORX=1 TO 12
1100 PRINTLEFT$(P$, 11) SPC(T3+X) "
    [<+>]"
1110 FORJ=1 TO 5: NEXT
1120 PRINTLEFT$(P$, 11) SPC(T3+X) "
    ";
1130 GETG$: GOSUB 3030: IFG$="" THEN 1100
1140 IFG$=CHR$(13) THEN POKE 198, 0: GOTO 1180
1150 PRINTLEFT$(P$, 11) SPC(T3+X) G$
1160 A$(Q, 3)=A$(Q, 3)+G$
1170 NEXT
1180 IFLC=1 THEN LC=0: GOTO 1630
1190 FORJ=1 TO 100: NEXT: GOSUB 270
1200 PRINTLEFT$(P$, 2) SPC(11) "
    [<4>]{ 4 SPAZI}PROVINCIA?

```

```

    { 4 SPAZI}{BLK}"
1210 FORX=1 TO 2
1220 PRINTLEFT$(P$, 14) SPC(T5+X) "
    [<+>]"
1230 FORJ=1 TO 5: NEXT
1240 PRINTLEFT$(P$, 14) SPC(T5+X) "
    ";
1250 GETG$: GOSUB 3030: IFG$="" THEN 1220
1260 IFG$=CHR$(13) THEN POKE 198, 0: GOTO 1300
1270 PRINTLEFT$(P$, 14) SPC(T5+X) G$
1280 A$(Q, 4)=A$(Q, 4)+G$
1290 NEXT
1300 IFLC=1 THEN LC=0: GOTO 1630
1310 FORJ=1 TO 100: NEXT: GOSUB 270
1320 PRINTLEFT$(P$, 2) SPC(11) "
    [<4>] CODICE POSTALE?
    { 2 SPAZI}{BLK}"
1330 FORX=1 TO 5
1340 PRINTLEFT$(P$, 14) SPC(T4+X) "
    [<+>]"
1350 FORJ=1 TO 5: NEXT
1360 PRINTLEFT$(P$, 14) SPC(T4+X) "
    ";
1370 GETG$: GOSUB 3030: IFG$="" THEN 1340
1380 PRINTLEFT$(P$, 14) SPC(T4+X) G$
1390 A$(Q, 5)=A$(Q, 5)+G$
1400 NEXT
1410 IFLC=1 THEN LC=0: GOTO 1630
1420 FORJ=1 TO 100: NEXT: GOSUB 270
1430 PRINTLEFT$(P$, 2) SPC(11) "
    [<4>]{ 4 SPAZI}TELEFONO?
    { 5 SPAZI}{BLK}"
1440 FORX=1 TO 4
1450 PRINTLEFT$(P$, 17) SPC(T5+X) "
    [<+>]"
1460 FORJ=1 TO 5: NEXT
1470 PRINTLEFT$(P$, 17) SPC(T5+X) "
    ";
1480 GETG$: GOSUB 3030: IFG$="" THEN 1450
1490 PRINTLEFT$(P$, 17) SPC(T5+X) G$
1500 C$=C$+G$
1510 NEXT
1520 FORX=1 TO 8
1530 PRINTLEFT$(P$, 17) SPC(T5+5+X) "
    [<+>]"
1540 FORJ=1 TO 5: NEXT
1550 PRINTLEFT$(P$, 17) SPC(T5+5+X) " "
    ";
1560 GETG$: GOSUB 3030: IFG$="" THEN 1530
1570 PRINTLEFT$(P$, 17) SPC(T5+5+X) G$
1580 C1$=C1$+G$
1590 NEXT
1600 A$(Q, 6)=C$+"/" + C1$: C$="": C1$=""

```



```

1610 IFLC=1THENLC=0:GOTO1630
1620 FORJ=1TO100:NEXT:GOSUB270
1630 FORJ=1TO100:NEXT:X=0:GOSUB270
1640 X=X+1:PRINTLEFT$(P$,2)SPC(10)"[<4>]TUTTO GIUSTO
      { 2 SPAZI}(S/N)?:IFX=3THEN
      X=0:GOTO1680
1650 FORJ=1TO500:NEXT:PRINTLEFT$(P$,2)SPC(10)"{ 20 SPAZI}"
1660 FORJ=1TO100:NEXT
1670 GOTO1640
1680 GETG$:IFG$=""THEN1680
1690 IFG$="S"THENGOSUB210:GOTO390
1700 IFG$="N"THEN1720
1710 GOTO1680
1720 FORJ=1TO100:NEXT:X=0:GOSUB270
1730 X=X+1:PRINTLEFT$(P$,2)SPC(10)"[<4>]{ 2 SPAZI}DATO ERRA
      TO N.?( 3 SPAZI)":IFX=3THEN
      X=0:GOTO1770
1740 FORJ=1TO500:NEXT:PRINTLEFT$(P$,2)SPC(10)"{ 20 SPAZI}"
1750 FORJ=1TO100:NEXT
1760 GOTO1730
1770 GETG$:IFG$=""THEN1770
1780 ONVAL(G$)GOTO1800,1830,1860,1890,1920,1950
1790 GOTO1770
1800 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]{ 2 SPAZI}CORREGGI NOM
      E{ 3 SPAZI}":A$(Q,1)=""
1810 FORZ=1219TO1195STEP-1:POKEZ,32:FORJ=1TO100:NEXT:NEXT
1820 LC=1:GOTO830
1830 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]CORREGGI INDIRIZZO":A$(Q,2)=""
1840 FORZ=1337TO1320STEP-1:POKEZ,32:FORJ=1TO100:NEXT:NEXT
1850 LC=1:GOTO950
1860 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]CORREGGI CITTA'
      { 2 SPAZI}":A$(Q,3)=""
1870 FORZ=1451TO1440STEP-1:POKEZ,32:FORJ=1TO100:NEXT:NEXT
1880 LC=1:GOTO1070
1890 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]CORREGGI PROVINCIA":A$(Q,4)=""
1900 FORZ=1559TO1558STEP-1:POKEZ,32:FORJ=1TO100:NEXT:NEXT
1910 LC=1:GOTO1190
1920 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]{ 3 SPAZI}CORREGGI CAP
      { 3 SPAZI}":A$(Q,5)=""
1930 FORZ=1577TO1573STEP-1:POKEZ,32:FORJ=1TO100:NEXT:NEXT
1940 LC=1:GOTO1310

```

```

1950 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]CORREGGI TELEFONO ":A$(Q,6)=""
1960 FORZ=1681TO1678STEP-1:POKEZ,32:FORJ=1TO100:NEXT:NEXT
1970 FORZ=1690TO1683STEP-1:POKEZ,32:FORJ=1TO100:NEXT:NEXT
1980 LC=1:GOTO1420
1990 GOTO1770
2000 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]CARICAMENTO DATI ":FORJ=1TO2000:NEXT
2010 IFDV$="D"THEN2060
2020 PRINT"{CLR}{ 3 GIU'}{RED}
      { 2 SPAZI}INSERISCI LA CASS
      ETTA-DATI RIAVVOLTA"
2030 PRINTSPC(12)"E PREMI UN TAS
      TO"
2040 GETG$:IFG$=""THEN2040
2050 OPEN7,1,0,"DATI":GOTO2100
2060 PRINTLEFT$(P$,2)SPC(11)"
      { 2 SPAZI}ATTENDI PREGO
      { 3 SPAZI}"
2070 OPEN7,8,7,"DATI,S,R"
2080 OPEN15,8,15:INPUT#15,E$,E1$,E2$,E3$
2090 IFVAL(E$)=62THENCLOSE7:CLOSE15:GOTO390
2100 SS=0
2110 SS=SS+1
2120 FORJ=1TO6:INPUT#7,A$(SS,J):
      NEXT
2130 IFST=0THEN2110
2140 Q=SS
2150 CLOSE7:CLOSE15
2160 GOTO390
2170 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]SALVATAGGIO DATI ":FORJ=1TO2000:NEXT
2180 IFDV$="D"THEN2230
2190 PRINT"{CLR}{ 3 GIU'}{RED}
      { 2 SPAZI}INSERISCI LA CASS
      ETTA-DATI RIAVVOLTA"
2200 PRINTSPC(12)"E PREMI UN TAS
      TO{GIU'}"
2210 GETG$:IFG$=""THEN2210
2220 OPEN6,1,1,"DATI":GOTO2250
2230 PRINTLEFT$(P$,2)SPC(11)"
      { 2 SPAZI}ATTENDI PREGO
      { 3 SPAZI}"
2240 OPEN6,8,6,"@:DATI,S,W"
2250 FORLS=1TOQ
2260 FORJ=1TO6:PRINT#6,A$(LS,J):
      NEXT:NEXT
2270 CLOSE6
2280 GOTO3000
2290 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]{ 2 SPAZI}CANCELLAZION
      E{ 3 SPAZI}":FORJ=1TO1000:N
      EXT

```



```

2300 FORJ=1TO100:NEXT:GOSUB270
2310 PRINTLEFT$(P$,2)SPC(11)"
      { 6 SPAZI}NOME?{ 7 SPAZI}
      {BLK}"
2320 FORX=1TO25
2330 PRINTLEFT$(P$,5)SPC(T2+X)"
      [<+>]"
2340 FORJ=1TO5:NEXT
2350 PRINTLEFT$(P$,5)SPC(T2+X)"
      ";
2360 GETG$:GOSUB3030:IFG$=""THEN
      2330
2370 IFG$=CHR$(13)THENPOKE198,0:
      GOTO2410
2380 PRINTLEFT$(P$,5)SPC(T2+X)G$

2390 DL$=DL$+G$
2400 NEXT
2410 FORAS=1TOQ:IFDL$=A$(AS,1)TH
      EN2460
2420 NEXT
2430 PRINTLEFT$(P$,2)SPC(11)"
      [<4>] NOME NON TROVATO ":FO
      RJ=1TO2000:NEXT
2440 DL$=""
2450 GOTO390
2460 DL$="":SD=0
2470 Q=Q-1
2480 FORSD=ASTOQ
2490 FORJ=1TO6:A$(SD,J)=A$(SD+1,
      J):NEXT:NEXT
2500 GOTO390
2510 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]{RVS}V{OFF}IDEO O
      {RVS}S{OFF}TAMPANTE {BLK}"
2520 GETG$:IFG$=""THEN2520
2530 IFG$="V"THEN2560
2540 IFG$="S"THEN2750
2550 GOTO2520
2560 FORER=1TOQ
2570 IFA$(ER,1)=""THEN2730

2580 PRINTLEFT$(P$,5)SPC(T2+1)A$
      (ER,1)
2590 PRINTLEFT$(P$,8)SPC(T3+1)A$
      (ER,2)
2600 PRINTLEFT$(P$,11)SPC(T3+1)A
      $(ER,3)
2610 PRINTLEFT$(P$,14)SPC(T5+1)A
      $(ER,4)
2620 PRINTLEFT$(P$,14)SPC(T4+1)A
      $(ER,5)
2630 PRINTLEFT$(P$,17)SPC(T5+1)A
      $(ER,6)

2640 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]{ 3 SPAZI}PREMI UN TAS
      TO{ 3 SPAZI}{BLK}"
2650 GETG$:IFG$=""THEN2650
2660 FORZ=1219TO1195STEP-1:POKEZ
      ,32:FORJ=1TO10:NEXT:NEXT
2670 FORZ=1337TO1320STEP-1:POKEZ
      ,32:FORJ=1TO10:NEXT:NEXT

2680 FORZ=1451TO1440STEP-1:POKEZ
      ,32:FORJ=1TO10:NEXT:NEXT
2690 FORZ=1559TO1558STEP-1:POKEZ
      ,32:FORJ=1TO10:NEXT:NEXT
2700 FORZ=1577TO1573STEP-1:POKEZ
      ,32:FORJ=1TO10:NEXT:NEXT
2710 FORZ=1681TO1678STEP-1:POKEZ
      ,32:FORJ=1TO10:NEXT:NEXT
2720 FORZ=1690TO1683STEP-1:POKEZ
      ,32:FORJ=1TO10:NEXT:NEXT
2730 NEXT
2740 GOTO390
2750 PRINT"{CLR}{ 3 GIU'}{RED}
      { 2 SPAZI}POSIZIONA LA CART
      A AD INIZIO PAGINA"

2760 PRINTSPC(12)"E PREMI UN TAS
      TO"
2770 GETG$:IFG$=""THEN2770
2780 OPEN4,4:CMD4
2790 GOSUB3010
2800 FORPR=1TOQ
2810 PL=PL+1
2820 IFA$(PR,1)=""THEN2890

2830 PRINTA$(PR,1)TAB(26-LEN(A$(
      PR,1)))A$(PR,2);
2840 PRINTTAB(19-LEN(A$(PR,2)))A
      $(PR,3);
2850 PRINTTAB(13-LEN(A$(PR,3)))A
      $(PR,4);
2860 PRINTTAB(3-LEN(A$(PR,4)))A$
      (PR,5);
2870 PRINTTAB(6-LEN(A$(PR,5)))A$
      (PR,6);

2880 PRINT
2890 IFPL=60THENFORPD=1TO4:PRINT
      :NEXT:GOSUB3010
2900 NEXT:IFPL=0THEN2920
2910 PD=62-PL:FORPK=1TOPD:PRINT:
      NEXT
2920 PRINT#4:CLOSE4
2930 GOTO390
2940 IFQ=0THEN3000
2950 PRINTLEFT$(P$,2)SPC(11)"
      [<4>]SEI SICURO{ 2 SPAZI}(S
      /N)?:POKE53247,3

2960 GETEN$:IFEN$=""THEN2960
2970 IFEN$="S"THEN2170
2980 IFEN$="N"THEN390
2990 GOTO2960
3000 PRINT"{CLR}":SYS64738
3010 PRINT"NOME"SPC(22)"INDIRIZZ
      O"SPC(10)"CITTA'"SPC(7)"PR
      CAP{ 3 SPAZI}TELEFONO"
3020 FORL=0TO79:PRINT"=";:NEXT:P
      L=0:RETURN
3030 IFG$="{GIU'}"ORG$="{DES}"OR
      G$="{SU}"ORG$="{SIN}"ORG$=C
      HR$(20)THENG$=""
3040 RETURN

```


Dots

Dots è un programma educativo creato per insegnare ai bambini in età prescolare a distinguere numeri e lettere. Essi potranno far comparire un disegno, scelto tra quattro possibili, all'interno di una matrice in cui ogni casellina è indicata dalle sue coordinate (numeri per le righe e lettere per le colonne).

I disegni vengono tracciati congiungendo per mezzo di linee determinate caselle della matrice, nelle quali appaiono di volta in volta dei pallini. Ogni pallino aggiunge una linea al disegno soltanto se il bambino batte sulla tastiera le coordinate esatte, e via di seguito finché la figura non è completa.

```

1 PRINT"{CLR}{ 3 SPAZI}{ 2 Q}
  { 3 SPAZI}{ 2 Q}{ 2 SPAZI}
  { 3 Q}{ 2 SPAZI}{ 2 Q}
  { 6 SPAZI}Q{SPAZI}Q{SPAZI}Q
  { 2 SPAZI}Q{ 2 SPAZI}Q
  { 2 SPAZI}Q{ 8 SPAZI}Q{SPAZI}
  Q{SPAZI}Q{ 2 SPAZI}Q
  { 2 SPAZI}Q{ 3 SPAZI}Q
2 PRINT"{ 3 SPAZI}Q{SPAZI}Q
  {SPAZI}Q{ 2 SPAZI}Q{ 2 SPAZI}
  Q{ 4 SPAZI}Q{ 6 SPAZI}{ 2 Q}
  { 3 SPAZI}{ 2 Q}{ 3 SPAZI}Q
  { 2 SPAZI}{ 2 Q}
3 A$="Z":POKE36878,15:GOSUB910
6 PRINT"{ 5 SPAZI}{ 2 GIU'}{RVS}
  {PUR}ISTRUZIONI":PRINT"{GIU'}C
  ERCA IL PUNTO TONDO.
7 PRINT"{GIU'}BATTI IL NUMERO DE
  LLA RIGA E 'RETURN'.
  { 6 SPAZI}{GIU'}BATTI LA LETTE
  RA DELLA";
8 PRINT"COLONNA E 'RETURN'.":PRI
  NT"{ 2 GIU'}{BLK}PREMI UN TAST
  O.
9 GETS$:IFSS$=""THEN9
10 PRINT"{CLR}{ 2 GIU'}{RED}COSA
  VUOI DISEGNARE?":PRINT"{BLU}
  {GIU'}{RVS}A{OFF}=FARFALLA (2
  2 PUNTI)
11 PRINT"{GIU'}{RVS}B{OFF}=FUNGO
  (12 PUNTI)":PRINT"{GIU'}
  {RVS}C{OFF}=CANE (20 PUNTI)
12 PRINT"{GIU'}{RVS}D{OFF}=CUORE
  (10 PUNTI)
18 GETA$:IFA$=""THEN18
19 L=7746:C=38466:POKE36879,26:C
  D=30720
20 PRINT"{CLR}{ 2 GIU'}
  { 2 SPAZI}ABCDEFGHIJKLMNPOQR
30 FORT=1TO418:POKEL,79:L=L+1:PO
  KEC,3:C=C+1:NEXT

```

```

40 PRINT"{HOME}{ 21 GIU'}
  { 2 SPAZI}ABCDEFGHIJKLMNPOQR
  { 2 SPAZI}";
50 PRINT"{HOME}{ 3 GIU'} 1"SPC(1
  8)"1{ 2 SPAZI}2"SPC(18)"2
  { 2 SPAZI}3"SPC(18)"3
  { 2 SPAZI}4"SPC(18)"4
  { 2 SPAZI}5"SPC(18)"5 ";
60 PRINT" 6"SPC(18)"6{ 2 SPAZI}
  7"SPC(18)"7{ 2 SPAZI}8"SPC(18)
  )"8{ 2 SPAZI}9"SPC(18)"9 10"S
  PC(18)"1011";
65 PRINTSPC(18)"11"
70 PRINT"{SU}12"SPC(18)"1213"SPC
  (18)"1314"SPC(18)"1415"SPC(18)
  )"1516"SPC(18)"1617"SPC(18)
80 PRINT"1718"SPC(18)"18{HOME}"
89 RESTORE
90 READB$:IFB$=A$THEN92
91 GOTO90
92 IFA$="A"THEN500
93 IFA$="B"THEN550
94 IFA$="C"THEN600
95 IFA$="D"THEN650
99 FORT=1TO500:NEXT
100 READA,B,E$,S:IFA=0THEN900
101 POKEA,81:POKEA+CD,B
105 PRINT"{HOME}{ 43 SPAZI}{SU}"
110 INPUT"{BLU}{HOME}+NUMERO";N$
115 FORT=1TO500:NEXT
120 PRINTSPC(11)"{ 2 SU}↑LETTERA
  "":INPUTL$
130 IFE$=N$+L$THEN200
140 PRINT"{SU}{ 7 SPAZI}RIPROVA"
  :POKE36874,209:FORT=1TO500:N
  EXT:POKE36874,195
150 FORT=1TO500:NEXT:POKE36874,0
160 FORT=1TO1500:NEXT:GOTO105

```



```

200 FORT=1TO1000:NEXT:FORT=1TO18
   :POKESQ,160:POKESQ+CD,B:IFSQ
   =ATHEN220
210 SQ=SQ+S:NEXT
220 FORZ=200TO235:POKE36876,Z:NE
   XT:POKE36876,0:GOTO99
500 POKE7799,77:POKE7799+CD,0:PO
   KE7801,78:POKE7801+CD,0:POKE
   7822,160:POKE7822+CD,5
510 SQ=7822:GOTO99
550 POKE7976,160:POKE7976+CD,4:P
   OKE7977,160:POKE7977+CD,4:PO
   KE8058,160:POKE8058+CD,5
560 SQ=8058:GOTO99
600 POKE7903,74:POKE7903+CD,0:PO
   KE7904,75:POKE7904+CD,0:POKE
   7882,85:POKE7882+CD,0
610 POKE7783,160:POKE7783+CD,2:S
   Q=7783:GOTO99
650 PRINT"{ 8 GIU' } { 7 DES } { GRN }
   ELENA { 2 GIU' } { 5 SIN } { BLU } L
   AURA { 2 GIU' } { 5 SIN } { YEL } SI
   LVIA"
660 POKE7822,160:POKE7822+CD,2:S
   Q=7822:GOTO99
800 PRINT"{HOME} BUON LAVORO! AN
   CORA? "
810 Y$="":GETY$:IFY$=""THEN810
820 IFY$="S"THEN1
830 IFY$="N"THENPRINT"{CLR}
   { 8 GIU' } { 7 SPAZI } BYE NOW!"
   :FORT=1TO3000:NEXT:END
840 GOTO810
900 PRINT"{HOME} BUON LAVORO! AN
   CORA? "
910 READB$:IFB$="Z"THEN930
920 GOTO910
930 READP,D:IFP=-1ANDAS="Z"THENP
   OKE36875,0:RETURN
940 IFP=-1THENPOKE36875,0:GOTO96
   0
950 POKE36875,P:FORT=1TOD:NEXT:G
   OTO930
960 Y$="":GETY$:IFY$=""THEN960
970 IFY$="S"THEN10
980 IFY$="N"THENPRINT"{CLR}
   { 8 GIU' } { 7 SPAZI } CIAO!!!":
   FORT=1TO3000:NEXT:END

```

```

990 GOTO960
1000 DATAA,7888,5,7I,22,7804,2,3
   M,-21
1010 DATA7806,2,30,1,7852,2,5Q,2
   3,7940,2,9Q,22,7982,2,110,2
   1,8028,7,13Q,23
1020 DATA8072,7,15Q,22,8093,7,16
   P,21,8090,7,16M,-1,7998,7,1
   2I,-23,8082,7,16E,21
1030 DATA8079,7,16B,-1,8056,7,15
   A,-23,8012,7,13A,-22,7970,7
   ,11C,-21,7924,2,9A,-23
1040 DATA7836,2,5A,-22,7794,2,3C
   ,-21,7796,2,3E,1,7888,2,7I,
   23,8042,5,14I,22,0,0,0,0
1500 DATAB,8063,5,15H,1,7975,4,1
   1H,-22,7972,4,11E,-1,7950,4
   ,10E,-22,7908,4,8G,-21
1510 DATA7913,4,8L,1,7959,4,10N,
   23,7981,4,11N,22,7978,4,11K
   ,-1,8066,4,15K,22
1520 DATA8071,5,15P,1,8063,5,15H
   ,-1,0,0,0,0
2000 DATAC,7867,2,6J,21,7862,2,6
   E,-1,7883,2,7D,21,7993,2,12
   D,22,8039,2,14F,23
2010 DATA8040,2,14G,1,8018,2,13G
   ,-22,7995,2,12F,-23,7973,2,
   11F,-22,7977,2,11J,1
2020 DATA8046,2,14M,23,8047,2,14
   N,1,8003,2,12N,-22,7980,2,1
   1M,-23,7892,2,7M,-22
2030 DATA7850,2,5O,-21,7852,0,5Q
   ,1,7830,0,4Q,-22,7829,0,4P,
   -1,7783,2,2N,-23,0,0,0,0
2500 DATAD,7753,2,1F,-23,7750,2,
   1C,-1,7792,2,3A,21,7924,2,9
   A,22,8108,2,17I,23
2510 DATA7940,2,9Q,-21,7808,2,3Q
   ,-22,7762,2,10,-23,7759,2,1
   L,-1,7822,2,4I,21,0,0,0,0
4000 DATAZ,225,300,231,100,235,2
   00,240,200,235,200,231,200,
   225,200,231,200,235,300
4010 DATA231,100,225,200,215,200
   ,225,100,0,300,225,100,-1,0

```



Crash

In questo gioco devi guidare la tua auto lungo una pista composta da quattro circuiti concen-

trici; al tuo passaggio cancelli i puntini disegnati sulla strada. Quando li avrai cancellati tutti,

potrai passare alla pista successiva. Attenzione, però: devi evitare di scontrarti con un'auto rivale, che percorre anch'essa il circuito ma in direzione opposta alla tua. Puoi passare da un circuito a quello adiacente attraversando le "porte" costituite dalle aree a scacchiera; per far questo, devi pilotare l'auto

per mezzo dei tasti F5 (su), F7 (giù), S (sinistra) e D (destra). Ogni nuova pista completa frutta più punti della precedente; sullo schermo di gioco vengono mostrati il numero della pista, le migliaia di punti ottenuti, le auto ancora a disposizione (tre inizialmente) ed il record. Questo programma gira sul VIC inespanso.

```

0 POKE36879,9:POKE36878,15:PRINT
  "{CLR}{WHT}{GIU'}":GOSUB700:MO
  =4
1 TH=0:MO=MO-1:IFMO<0THENMO=2
2 A=-1:B=22:C=1:D=-22:S=7756:Z=0

3 K=-1:L=-22:M=1:N=22:R=8117
10 PO=PO+1:C1=0:PRINT"{CLR}
  { 2 GIU'}{WHT}{<A>}{ 19 * }
  [<S>]"
11 PRINT"-. . . . . -"
12 PRINT"-. [<A>]{ 5 * } [<+>]*
  {RVS} {OFF}* [<+>]{ 5 * } [<S>].
  -"
13 PRINT"-. . . . . -"
14 PRINT"-. . . [<A>]{ 3 * } [<+>]*
  {RVS} {OFF}* [<+>]{ 3 * } [<S>].
  - . -"
15 PRINT"-. . . . . -"
16 PRINT"-. . . . . [<A>]* [<+>]* {RVS}
  {OFF}* [<+>]* [<S>]. - . -"
17 PRINT"-. . . . . -"
18 PRINT"-. . . . . [<+>]. [<+>]. [<+>].
  [<A>]{ 3 * } [<S>]. [<+>]. [<+>].
  [<+>]. -"
19 PRINT"-. {RVS} {OFF}. {RVS}
  {OFF}. {RVS} {OFF}. -
  { 3 SPAZI}-. {RVS} {OFF}. {RVS}
  {OFF}. {RVS} {OFF}. -"
20 PRINT"-. [<+>]. [<+>]. [<+>].
  [<Z>]{ 3 * } [<X>]. [<+>]. [<+>].
  [<+>]. -"
21 PRINT"-. . . . . -"
22 PRINT"-. . . . . [<Z>]* [<+>]* {RVS}
  {OFF}* [<+>]* [<X>]. - . -"
23 PRINT"-. . . . . -"
24 PRINT"-. . . . . [<Z>]{ 3 * } [<+>]*
  {RVS} {OFF}* [<+>]{ 3 * } [<X>].
  - . -"
25 PRINT"-. . . . . -":
  PRINT"-. [<Z>]{ 5 * } [<+>]*
  {RVS} {OFF}* [<+>]{ 5 * } [<X>].
  -"
26 PRINT"-. . . . . -":
  PRINT" [<Z>]{ 19 * } [<X>]"
27 T=PEEK(S):CH=CH+1:IFCH>1THENG
  OSUB600
28 PRINT"{HOME}{ 21 GIU'}RECORD:
  "HI:Z=Z+1:IFZ<2THEN41
29 PRINT"{HOME}{GIU'}PISTA:"PO;T
  AB(11)TH;:FORI=7720TO7722-ED

```

```

30 POKEI,87:NEXT:FORI=7723-EDTO7
  723:POKEI,32:NEXT:Z=0:IFPEEK(
  S+A)=46THEN33
31 IFPEEK(S+A)=87THEN500
32 IFPEEK(S+A)<>32THENE=A:A=B:B=
  C:C=D:D=E
33 IFPEEK(S-22)=160THENA1=-44
34 IFPEEK(S+22)=160THENA1=44
35 IFPEEK(S-1)=160THENA1=-2
36 IFPEEK(S+22)=160ANDPEEK(S-22)
  =160THENA1=44*(INT(RND(1)*-3)
  +2)
37 IFPEEK(S+1)=160THENA1=2
38 IFPEEK(S+1)=160ANDPEEK(S-1)=1
  60THENA1=2*(INT(RND(1)*-3)+2)
39 IFABS(A1)=3ORABS(A1)=66THENA1
  =0
40 POKES,T:S=S+A+A1:T=PEEK(S):PO
  KES,81:A1=0:IFT=87THEN500
41 X=0:GETA$:IFA$=""THEN45
42 IFA$="{F5}"ANDABS(K)=1THENX=-
  44:V=-22
43 IFA$="{F7}"ANDABS(K)=1THENX=4
  4:V=22
44 IFA$="S"ANDABS(K)=22THENX=-2:
  V=-1
45 IFA$="D"ANDABS(K)=22THENX=2:V
  =1
46 IFPEEK(R+V)<>102THENX=0
47 SO=SO+2:POKER,32:IFPEEK(R+K+X
  )=46THENSC=SC+PO:POKE36876,18
  0:C1=C1+1:GOTO51
48 IFX<>0THEN51
49 IFPEEK(R+K)=81THEN500
50 IFPEEK(R+K)<>32THENP=K:K=L:L=
  M:M=N:N=P:SO=0
51 S2$=STR$(SC):R=R+K+X:POKER,87
  :SC$="000"+RIGHT$(S2$,LEN(S2$
  )-1)
53 PRINT"{HOME}{ 11 GIU'}
  { 9 DES}"RIGHT$(SC$,3)
55 X=0:IFC1=160THEN1
56 IFSC>999THENTH=TH+1:SC=0
57 IFSC+(TH*1000)>HITHENHI=SC+(T
  H*1000)
60 POKE36876,0:GOTO28
500 POKER,86:SO=0
503 FORI=230TO128STEP-5:FORO=1TO
  3:POKE36873+O,I:NEXTO,I

```



```

504 FORI=1TO4:POKE36873+I,0:NEXT
:LS$=SC$
505 ED=ED+1:IFED=3THENED=0:PRINT
"{HOME}{ 5 GIU' }"TAB(6)GM$:F
ORI=1TO6000:NEXT:PRINT"{CLR}
":GOTO0
507 POKER,32:POKES,T:IFT=87THENP
OKES,32
508 A=-1:B=22:C=1:D=-22:S=7756:Z
=0:K=-1:L=-22:M=1:N=22:R=811
7
509 T=PEEK(S):FORI=1TO500:NEXT:G
OTO28
600 POKE36877,0:FORI=7724TO8142
601 IFPEEK(I)=102THENPOKEI,160:P
OKE36876,222:FORO=1TO10:NEXT
:GOTO603
602 IFPEEK(I)=160THENPOKEI,102:P
OKE36876,250:FORO=1TO10:NEXT
603 POKE36876,0:NEXT:RETURN
700 E$="{CLR}{GIU'}{ 2 SPAZI}PRO
NTO A GIOCARE A{ 3 SPAZI}":F
ORI=1TO23:PRINTMID$(E$,I,1);
:CH=0
701 IFMID$(E$,I,1)<>" "THENPOKE3
6875,235
702 FORO=1TO99:NEXT:POKE36875,0:
NEXT:B$(1)="{ 2 DES}{GIU'}
{PUR}N[< 2 +>]{GIU'}
{ 3 SIN}[<+>]{GIU'}{SIN}M
[< 2 +>]{DES}":Z=0:SC=0:C1=0
:PO=0
703 B$(2)="{WHT}[< 2 +>]M{GIU'}
{ 3 SIN}[<+>]L[<+>]{GIU'}
{ 3 SIN}[<+>]{ 2 M}{DES}"
704 B$(3)="{CYN}N[<+>]M{GIU'}
{ 3 SIN}[<+>][<E>][<+>]
{GIU'}{ 3 SIN}[<+>]{DES}
[<+>]{DES}"
705 B$(4)="{DES}{WHT}N[< 2 +>]
{GIU'}{ 3 SIN}{ 2 M}{GIU'}
{ 3 SIN}[< 2 +>]N{DES}"
706 B$(5)="{PUR}[<+>]{DES}[<+>]
{GIU'}{ 3 SIN}[< 3 +>]{GIU'}
{ 3 SIN}[<+>]{DES}[<+>]
{GIU'}{ 5 DES}"
707 D$="PREMI {RVS}RETURN{OFF}":
GM$="{GRN}{RVS}FINE GIOCO"
708 SC$=STR$(VAL(RIGHT$(SC$,3))+
TH*1000):IFSC$=""THENSC$="00
0"
709 C1$="{GIU'}{WHT}PUNTEGGIO:"+
SC$:FORI=1TO10
710 POKE36876,150+(I*5):POKE8054
+I,81:POKE8075-I,87:FORO=1TO
100:NEXT:POKE8054+I,32
711 POKE8075-I,32:NEXT:R=8054+I:
POKER,86:POKER-1,81:POKE3687
6,0
715 FORI=1TO5:PRINTB$(I);:NEXT:P
OKE36877,200:FORI=15TO0STEP-
.05:POKE36878,I:NEXT

```

```

720 POKE36878,15:POKE36877,0
731 FORI=1TOLEN(C$):PRINTMID$(C$
,I,1);:IFMID$(C$,I,1)<>" "TH
ENPOKE36875,235
733 FORO=1TO100:NEXT:POKE36875,0
:NEXT:PRINT"{ 2 GIU' }"
735 FORI=1TOLEN(C1$):PRINTMID$(C
1$,I,1);:IFMID$(C1$,I,1)<>"
"THENPOKE36875,235
736 FORO=1TO100:NEXT:POKE36875,0
:NEXT:PRINT"{ 2 GIU' }"
737 FORI=1TOLEN(D$):PRINT"{SU}"L
EFT$(D$,I):IFMID$(D$,I,1)<>"
"THENPOKE36875,235
738 FORO=1TO100:NEXT:POKE36875,0
:NEXT
740 FORI=1TO1000:GETA$:IFA$=""TH
EN750
745 IFASC(A$)=13THENRETURN
750 NEXT:PRINT"{CLR}{GIU'}{RVS}
{ 6 SPAZI}CONTROLLI:
{ 6 SPAZI}"
755 PRINT"{ 2 DES}[<@>]{GIU'}
{ 2 SIN}[<M>]S[<G>]{GIU'}
{ 2 SIN}[<T>]{DES}{SU}:SINIS
TRA{GIU'}"
756 PRINT"{ 2 DES}[<@>]{GIU'}
{ 2 SIN}[<M>]D[<G>]{GIU'}
{ 2 SIN}[<T>]{DES}{SU}:DESTR
A{GIU'}"
757 PRINT"{DES}[< 2 @>]{GIU'}
{ 3 SIN}[<M>]F5[<G>]{GIU'}
{ 3 SIN}[< 2 T>]{DES}{SU}:SU
{GIU'}"
758 PRINT"{DES}[< 2 @>]{GIU'}
{ 3 SIN}[<M>]F7[<G>]{GIU'}
{ 3 SIN}[< 2 T>]{DES}{SU}:GI
U'"
760 FORI=1TO6000:NEXT:RETURN

```



Disegnatore

Questo programma permette di creare disegni in alta risoluzione con il VIC inespanso, usando una matrice di 12x15 caratteri (corrispondenti a 96x120=11520 punti). I disegni possono essere salvati su nastro o su disco ed in seguito ricaricati.

Dopo aver digitato e salvato il programma, fatelo girare. La prima cosa che vi viene richiesta è di selezionare, per mezzo dei tasti numerici 1-8, i colori dello sfondo e del disegno; il programma predispone quindi lo schermo ad alta risoluzione, in cui compare un punto che serve da cursore. Se disponete di un joystick, con esso potete pilotare il cursore, che può spostarsi tracciando una linea (modo 1) oppure no (modo 2). A questi due modi di funzionamento corrispondendo due diversi segnali acustici; il programma inizia in modo 2, e per commutare tra i due modi si deve premere il pulsante del joystick. In modo 2 il cursore non solo

si sposta senza tracciare linee, ma cancella i punti su cui passa.

Se invece non possedete un joystick, dovete modificare leggermente il programma. Cancellate le linee 330 e 340, e cambiate la linea 320: 320 A=PEEK(197):J0=-(A=44):J1=-(A=36):J2=-(A=20):J3=-(A=12):IF A=32THENB=ABS(B-1)

In questo modo il movimento del cursore viene comandato dai tasti I (su), M (giù), J (sinistra) e K (destra); per commutare tra i due modi si usa la barra spaziatrice.

Tre ulteriori opzioni vengono offerte dai tasti funzione. Con F1 si cancella lo schermo ad alta risoluzione senza spostare il cursore; con F3 il programma ricomincia dall'inizio (come con il RUN), permettendo di caricare da nastro o da disco un disegno salvato precedentemente; infine con F5 si accede alla routine che salva su nastro o su disco il disegno eseguito.

```

10 POKE36869,240:POKE52,24:POKE5
  6,24:POKE36879,27:CLR
40 PRINT"{CLR}{ 3 GIU' }1) CREA U
  N NUOVO DISE-GNO."
50 PRINT"{GIU' }2) CARICA UN DISE
  GNO{ 2 SPAZI}IN MEMORIA.":POK
  E198,0
60 GETA$:IFA$<>"1"ANDA$<>"2"THEN
  60
70 IFA$="2"THEN450
80 PRINT"{CLR}{ 3 GIU' }COLORE DE
  L BORDO?":GOSUB100:G=VAL(A$)-
  1
90 PRINT"{ 3 GIU' }COLORE DELLO S
  FONDO?":GOSUB100:H=VAL(A$):GO
  TO120
100 GETA$:IFA$<"1"ORA$>"8"THEN10
  0
110 RETURN
120 POKE36879,G+16*H-8:PRINT"
  {CLR}"
130 FORX=0TO21:FORY=0TO22:POKE76
  80+X+22*Y,160:POKE38400+X+22
  *Y,G:NEXT:NEXT
140 IFW=1THEN160
  
```

```

150 FORI=6144TO7679:POKEI,0:NEXT
  :FORI=7424TO7431:POKEI,255:N
  EXT
160 C=0:POKE36869,254:FORX=5TO16
  :FORY=3TO18:IFX+Y=34THENNEXT
  :GOTO180
170 C=C+1:POKE7680+X+22*Y,C:NEXT
  :NEXT
180 Y=18:FORX=3TO18:POKE7680+X+2
  2*Y,160:NEXT:C=90
190 IFJ1THENF=F+1:IFF>7AND(C+1)/
  16<>INT((C+1)/16)THENF=0:C=C
  +1:GOTO210
200 IFF>7THENF=7
210 IFJ3THENF=F-1:IFF<0AND(C-1)/
  16<>INT((C-1)/16)THENF=7:C=C
  -1:GOTO230
220 IFF<0THENF=0
230 IFJ0THENE=E+1:IFE>7ANDC<177T
  HENE=0:C=C+16:GOTO250
240 IFE>7THENE=7
250 IFJ2THENE=E-1:IFE<0ANDC>16TH
  ENE=7:C=C-16:GOTO270
260 IFE<0THENE=0
  
```



```

270 POKE6144+8*C+F,PEEK(6144+8*C
+F)ANDNOT(2↑(7-E))
280 POKE6144+8*C+F,PEEK(6144+8*C
+F)OR2↑(7-E)
290 POKE36878,15:POKE36874+2*B,1
30+INT(C/2.14):POKE36878,0
300 POKE36874+2*ABS(B-1),0
310 IFB=0THENPOKE6144+8*C+F,PEEK
(6144+8*C+F)-2↑(7-E)
320 POKE37154,127:Z=128ANDPEEK(3
7152):J0=- (Z=0)
330 POKE37154,255:Z=PEEK(37151)
340 J1=- ((ZAND8)=0):J2=- ((ZAND16
)=0):J3=- ((ZAND4)=0):J=- ((ZA
ND32)=0):IFJTHENB=ABS(B-1)
350 GETA$:IFA$<>"{F1}"ANDA$<>"
{F3}"ANDA$<>"{F5}"THEN190
360 IFA$="{F3}"THEN10
370 IFA$="{F5}"THEN390
380 FORI=6144TO7423:POKEI,0:NEXT
:FORI=7432TO7679:POKEI,0:NEX
T:GOTO190
390 POKE36869,240:POKE36879,27
400 PRINT"{CLR}{GIU'}{RVS}N{OFF}
ASTRO O {RVS}D{OFF}ISCO?
{GIU'}":POKE198,0
402 GETA$:IFA$<>"N"ANDA$<>"D"THE
N402
405 IFA$="N"THENPRINT"RIAVVOLGER
E IL NASTRO"

```

```

415 PRINT"QUANDO SEI PRONTO
{ 5 SPAZI}PREMI (SPACE). "
420 B$="":GETB$:IFB$<>" "THEN420
425 IFA$="N"THENOPEN1,1,1
426 IFA$="D"THENINPUT"NOME DEL F
ILE";N$:N$=N$+","S,W":D=8:OPE
N1,8,5,N$
430 PRINT#1,G:PRINT#1,H:FORA=614
4TO7679:PRINT#1,PEEK(A):NEXT
:CLOSE1
440 PRINT"{CLR}IL DISEGNO E' SAL
VATO.":GOTO10
450 PRINT"{GIU'}{RVS}N{OFF}ASTRO
O {RVS}D{OFF}ISCO?":D=1:N$=
""
455 GETA$:IFA$<>"N"ANDA$<>"D"THE
N455
460 IFA$="N"THENPRINT"{ 3 GIU'}
INSERIRE E RIAVVOLGERELA CAS
SETTA"
470 PRINT"QUANDO SEI PRONTO
{ 5 SPAZI}PREMI (SPACE).
{GIU'}":W=1:WAIT198,1
475 IFA$="N"THENOPEN1,1,0
480 IFA$="D"THENINPUT"NOME DEL F
ILE";N$:D=8:N$=N$+","S,R":OPE
N1,8,5,N$
490 INPUT#1,G:INPUT#1,H:FORA=614
4TO7679:INPUT#1,C:POKEA,C:NE
XT:CLOSE1
500 GOTO120

```



Joystick

Fotoreporter

È stata scoperta nelle fogne di una grande città americana una razza di formiche mutanti di dimensioni colossali. Il 'Time' ti ha incaricato di effettuare un reportage fotografando una formica e tu hai accettato, attirato dalle forti possibilità di guadagno. Ti sei quindi procurato una cartina della rete fognaria e un aiutante che dalla superficie localizzerà le formiche per mezzo di un'apparecchiatura a raggi infrarossi.

Le fotografie riuscite della formica verranno pagate a seconda dell'inquadratura effettuata: 75 dollari per un'istantanea della testa, 25 per una laterale ed infine solo 10 per una scaden-

te foto da tergo. Se per caso dovessi dimenticarti di avanzare la pellicola, puoi vendere le doppie esposizioni ad un giornale di provincia, che però te le pagherà solamente un dollaro l'una.

La spesa del materiale è a tuo carico, perciò ogni foto sprecata o non scattata ti costa 2 dollari.

Sullo schermo del C64 appare la mappa della rete fognaria, sempre diversa ogni volta che si gioca, con indicata la posizione del giocatore e quella della formica; viene mostrato inoltre ciò che si vede nel mirino della macchina fotografica. Compagno anche un contasecondi,

l'indicatore di effettuato avanzamento pellicola e il contafotogrammi. Il funzionamento del joystick (porta 2) è leggermente diverso da quello usuale: spinto verso l'alto fa proseguire dritto il fotografo; verso il basso lo fa retrocedere, mentre a destra e sinistra lo fa ruotare di 90

gradi nelle rispettive direzioni. Il pulsante serve per scattare le foto ed infine la barra spaziatrice fa avanzare la pellicola. Il gioco termina quando tutti i 24 fotogrammi del rullino sono stati esposti, oppure quando il contasecondi indica che sono trascorsi 5 minuti.

```

0 DIMA%(5,6),B%(5,6),C%(5,6),O%(
5),Q%(4),M%(4):I=RND(-TI):GOTO
23
1 PRINTCS$(GIU')$INT(TI/60):IFTI
>TJGOTO59
2 Q=-2*(PEEK(T+O%(0))<M)-4*(PEEK
(T+O%(0-1))<M)-(PEEK(T+O%(0+1)
)<M)
3 FORI=0TO5:X=I+S:FORJ=0TO6:A=A%(
I,J):IFA=0GOTO7
4 IFJ=0THENIFPTHEMIF(AANDQ)=(AAN
DV)THENJ=6:GOTO7
5 Z=W*J+X:IFAANDQTHENPOKEZ,C%(I,
J):GOTO7
6 POKEZ,B%(I,J)
7 NEXTJ,I:V=Q:RETURN
8 A=PEEK(D):IFCTHENA=(A/4AND7)+(
8ANDPEEK(D+1)/16)+(A/2AND16)
9 IF(AAND31)=31THENRETURN
10 ONLOG(NTAAND31)/L2GOTO13,14,
16,48:I=T+O%(0)
11 IFPEEK(I)=HTHENPOKET,H:T=I:PO
KET,G
12 RETURN
13 I=T-O%(0):GOTO11
14 O=1-O*(O<4)
15 G=Q%(0):POKET,G:RETURN
16 O=O-1-4*(O=1):GOTO15
17 I=N+O%(L):IFPEEK(I)=HTHENPOKE
N,H:N=I:GOTO19
18 I=L:L=1+FNR(4):IFI=LGOTO18
19 POKEN,Q%(L):RETURN
20 GOSUB8:GETAS$:IFAS$<>"$THENPRIN
TCS${2 GIU'}${2 SPAZI}I$:PRI
NTBS$ +K":WD=1
21 GOSUB1:GOSUB17:IFT+O%(0)<>NTH
ENONR+1GOTO20:R=0:P=0:GOSUB2:
P=1:GOTO20
22 R=1:B=L-O-4*(O>L):FORY=0TO2:Z
=U+W*Y:FORX=0TO1:POKEX+Z,FNF(
B):NEXTX,Y:GOTO20
23 DEFNFX=INT(RND(1)*X):M=128
:DEFNFZ=ASC(MID$(F$,1+Z*6+
Y*2+X))
24 C=1:W=22:L=7680:D=37151:POKED
+3,127:K=36867
25 FORI=1TO18:D$=D$+" ":C$=C$+"
*":NEXT:A$=LEFT$(D$,W/2-6):B$
=LEFT$(A$,W/2-10)
26 PRINT{CLR}$:PRINTAS${BLK}U
{8 *}I$:PRINTAS${CYN}{<*>}
{RVS}{6 SPAZI}{OFF}E{BLK}$

```

```

27 PRINTAS${CYN}{<*>}{RVS}
{4 SPAZI}{OFF}E{BLK}$
28 PRINTAS${2 SPAZI}{CYN}
{4 SPAZI}{BLK}{2 SPAZI}$
[<Q>]+{2 *}{<W>}:FORI=1TO3
:PRINTAS${8 SPAZI}$[<Q>]
[<W>]{2 SPAZI}{<Q>}{<W>}:NE
XT
29 PRINTAS${2 SPAZI}{GRN}
{4 SPAZI}{BLK}{2 SPAZI}$
[<Q>]+{2 *}{<W>}:PRINTAS$
-{GRN}{RVS}E{4 SPAZI}{<*>}
{OFF}{BLK}$
30 PRINTAS${GRN}{RVS}E
{6 SPAZI}{<*>}{OFF}{BLK}$:P
RINTAS$J{8 *}K$:PRINTBS$
{RVS}U"C$ "I"
31 FORI=1TO9:PRINTBS${RVS}$-{BLK}
"D$"$:NEXT:PRINTBS${RVS}J"C$
"K{OFF}$
32 E=L+12*W+W/2-11:S=L+3*W+W/2-4
:O=1:H=160:GOSUB47:H=32:U=S+2
+2*W
33 R%=S+10+2*W:O%(1)=-W:O%(2)=-1
:O%(3)=W:O%(4)=1:O%(0)=1:O%(5)
)=-W
34 Q%(1)=-W-W:Q%(2)=W-1:Q%(3)=W+
1:Q%(4)=1-W:C$="{HOME}" +B$
35 POKEI,H:Z=6:FORY=FNR(3)TO5:A=
I+O%(Y):IFPEEK(A)<>160GOTO38
36 F=0:G=0:J=0:FORX=1TO4:F=F-(PE
EK(A+Q%(X))=H):IFPEEK(A+O%(X)
)=HTHENG=G+1:J=X-J
37 NEXT:IFF<4ANDG<2OR(G<3ANDF<1)
OR(G=2ANDABS(J)=2)THENZ=Y:Y=6
38 NEXT:IFZ<6THENI=I+O%(Z):V=V+1
:GOTO35
39 IFV<95THENGOSUB47:GOTO35
40 FORI=0TO6:A%(0,I)=1:A%(5,I)=4
:FORJ=0TO5:B%(J,I)=H:C%(J,I)=
160:NEXT:NEXT
41 FORI=1TO4:A%(I,1)=2:A%(I,5)=2
:NEXT:FORI=2TO3:FORJ=2TO4:A%(
I,J)=2:NEXT:NEXT
42 B%(0,0)=95:B%(5,0)=105:B%(0,6)
)=233:B%(5,6)=223:P=1:L2=LOG(
2):TJ=18000
43 C%(1,1)=95:C%(4,1)=105:C%(1,5)
)=233:C%(4,5)=223:FORZ=1TO24:
READA

```



```

44 F$=F$+CHR$(A+46+M):NEXT:Q%(1)
   =30:Q%(2)=60:Q%(3)=22:Q%(4)=6
   2:GOSUB47
45 N=I:FR=1
46 GOSUB18:V=0:F=0:G=Q%(0):GOSUB
   47:T=I:POKET,G:GOSUB57:TI$="0
   00000":GOTO21
47 I=E+2+W+FNR(17)+W*FNR(9):A=PE
   EK(I):ON-(A<>H)GOTO47:RETURN
48 ONWD+1GOTO54:IFT+O%(0)<>NTHEN
   A$="FOTO SPRECATA":F=0:I=H:J=
   0:GOTO55
49 A$="CLICK":B=L-O-4*(O>L)+1:F=
   B:M%(B)=M%(B)+1
50 FORY=0TO2:Z=R%+W*Y:FORX=0TO1:
   POKEX+Z,FNF(B-1)-M:NEXT:NEXT
51 IFMSTHENMS=0:POKEK,46:GOTO56
52 IFCANDFNR(9)=1THENMS=1:A$="MA
   PPA RUBATA":POKEK,24:GOTO56
53 POKEN,H:GOSUB47:N=I:GOTO56
54 A$="SOVRAIMPRESSIONE":FR=FR+1
   *(FR>0):DE=DE+1:M%(F)=M%(F)-1
   :F=0:I=66:J=16
55 FORY=0TO2:Z=R%+W*Y:FORX=0TO1:
   POKEX+Z,I+FNR(J):NEXT:NEXT
56 PRINTC$SPC(10-LEN(A$)/2)A$:FO
   RI=1TO500:NEXT:WD=0

```

```

57 PRINTC$D$:PRINT:PRINT:PRINTB$
   "{GIU'}"FR:PRINTB$"{ 3 SU}
   { 3 SPAZI}":PRINTB$"
   { 3 SPAZI}":FR=FR+1
58 IFFR<25THENRETURN
59 PRINT"{CLR}FINE GIOCO":PRINT:
   PRINT"'TIME' COMPRA LE FOTO":
   PRINT
60 I=M%(3):A=I*75:PRINT"TESTA:"I
   "A $75":IFCTHENPOKEK,46
61 J=M%(2)+M%(4):A=A+25*J:PRINT"
   LATO:"J"A $25":I=I+J
62 J=M%(1):A=A+J*10:PRINT"CODA:"
   J"A $10":PRINT:J=24-I-J
63 PRINT"MANCATE:"J"A -$2":PRINT
   :A=A-J*2:PRINT"SOVRAIMPRESSIO
   NI:"
64 PRINTDE"A $1":A=A+DE:PRINT:PR
   INT"TOTALE = $"A:POKE198,0:I=
   TI
65 GETA$:IFA$=""ANDTI<I+4000GOTO
   65
66 RUN:DATA27,39,39,27,61,69,39,
   39,35,47,,29,27,39,35,35,,,27
   ,27,47,35,28,,

```

Demo L/M



Il programma dimostra gli effetti prodotti sullo schermo da ben sei routine in linguaggio macchina. Quattro fra queste realizzano il classico scroll di un carattere verso il basso, l'alto, destra e sinistra. Le altre due producono invece effetti particolari: la prima, detta 'Clear' pulisce di colpo la colonna a sinistra sullo scher-

mo; utilizzata in abbinamento con le routine da un lato dello schermo di quanto esce dal lato opposto). L'ultima routine realizza invece un interessantissimo effetto-esplosione. Le 6 routine sono completamente rilocabili e possono dunque essere facilmente inserite nei vostri listati.

```

10 CLEAR USR "A"-201
20 CLS : PRINT AT 0,4;"Questo
   programma dimostra";AT 1,4;
   "una serie di corte routine
   s";AT 2,4;"in linguaggio ma
   cchina."
30 GO SUB 9000
35 CLS
40 PRINT ' INVERSE 1;"1)";
   INVERSE 0;" ESPLOSIONE: un
   a routine che realizza un
   interessante effetto"

```

```

50 FOR n=1 TO 4: RANDOMIZE
   USR EXPLODE: PAUSE 50:
   NEXT n
60 PRINT ' INVERSE 1;"2)";
   INVERSE 0;" SCROLL SINISTR
   A: sposta un carattere a
   sinistra, facendolo riappar
   ire dall'altro lato"
65 PAUSE 150
70 FOR n=1 TO 3: RANDOMIZE
   USR left: NEXT n

```



```

80 PAUSE 1: PAUSE 100
90 PRINT ' INVERSE 1;"3)";
  INVERSE 0;" SCROLL DESTRA
  : sposta un carattere a
  destra , facendolo riappar
  ire dall'altro lato"
95 PAUSE 150
100 FOR n=1 TO 3: RANDOMIZE
  USR right: NEXT n
110 PAUSE 1: PAUSE 100
120 PRINT ' INVERSE 1;"4)";
  INVERSE 0;" SCROLL ALTO
  : sposta un carattere in
  alto, pulendo le linee i
  n basso"
125 PAUSE 150
130 FOR n=1 TO 11: RANDOMIZE
  USR up: NEXT n
140 PAUSE 1: PAUSE 100
150 PRINT AT 7,0; INVERSE 1;"5)
  "; INVERSE 0;" SCROLL BASS
  O : sposta un carattere
  in basso, pulendo le linee
  in alto"
155 PAUSE 150
160 FOR n=1 TO 3: RANDOMIZE
  USR down: NEXT n
170 PAUSE 1: PAUSE 100
180 PRINT '"" INVERSE 1;"6)";
  INVERSE 0;" CLEAR : cance
  lla la colonna all'estrema
  sinistra;puo' essereutilizz
  ata con le routines di sc
  roll laterale per evitare i
  l wrap-around"

```

```

185 PAUSE 200
190 FOR n=1 TO 3: RANDOMIZE
  USR left: NEXT n: PAUSE 100
200 RANDOMIZE USR clear
210 FOR n=1 TO 32: RANDOMIZE
  USR left+USR clear: NEXT n
8999 STOP
9000 LET UP=USR "A"-200
9010 LET DOWN=UP+4
9020 LET LEFT=DOWN+55
9030 LET RIGHT=LEFT+12
9040 LET CLEAR=RIGHT+12
9050 LET EXPLODE=CLEAR+14
9060 FOR N=UP TO EXPLODE+25:
  READ A: POKE N,A: NEXT N
9070 DATA 205,254,13,201
9080 DATA 33,255,86,6,24,14,224,
  17,32,0,237,82,126,25,119,4
  3,13,32,247,14,32,17,32,7,2
  37,82,126,25,119,43,13,32,2
  47,16,226,33,0,64,17,224,0,
  14,8,6,32,54,0,35,16,251,25
  ,13,32,245,201
9090 DATA 33,1,64,17,0,64,1,254,
  23,237,176,201
9100 DATA 33,253,87,17,254,87,1,
  254,23,237,184,201
9110 DATA 33,0,64,17,32,0,6,192,
  54,0,25,16,251,201
9120 DATA 33,255,75,17,205,255,6
  2,255,211,254,68,4,16,254,6
  2,0,211,254,68,4,16,254,25,
  56,237,201
9130 RETURN

```

Per ricevere gli arretrati di **PAPERsoft** compilare il seguente tagliando:

J.soft Compilare ed inviare in busta chiusa a:
 via Rosellini 12 - 20142 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome _____

Cognome _____

Indirizzo _____

CAP _____ Città _____ Prov. _____

Desidero ricevere i seguenti numeri arretrati di


PAPERsoft _____

a L. 2.000 cad. per un totale di L. a mezzo

assegno allegato

contanti allegati

LIBRERIA



JACKSON

**A Milano,
in via Mascheroni 14.
La prima software
libreria italiana**

Un tempo si andava in libreria per il gusto della scoperta, per il piacere di esser informati sulle novità. Per incontrarsi, discutere, chiedere un consiglio al libraio-amico. Tutto questo è ancora possibile, per un prodotto assolutamente nuovo: libri e riviste di informatica italiani ed esteri, software, giochi.

Dove? Alla **Libreria JACKSON.**

La prima software - libreria italiana.

A Milano, in via Mascheroni 14.

Tel. 02-437385

Vieni a trovarci: ti aspettiamo.